
  v1.4.8



  
    
  
  Bug fixes


	Commanded.Event.ErrorHandler now keeps surounding failure context by @drteeth in https://github.com/commanded/commanded/issues/617


  
    
  
  v1.4.7



  
    
  
  Enhancements


	Application-wide event handler error handling by @drteeth in https://github.com/commanded/commanded/pull/605
	chore: remove asdf file by @yordis in https://github.com/commanded/commanded/pull/570
	chore: improve docs about aggregate version by @yordis in https://github.com/commanded/commanded/pull/608
	Update include_aggregate_version documentation by @TylerPachal in https://github.com/commanded/commanded/pull/609


  
    
  
  Bug fixes


	Fix flakey test by @drteeth in https://github.com/commanded/commanded/pull/599
	feat: default aggregate lifespan configuration by @yordis in https://github.com/commanded/commanded/pull/548
	Aggregate.handle_* now properly handles lifespans by @drteeth in https://github.com/commanded/commanded/pull/606
	Allow registration handle_call/cast callbacks to be called by @drteeth in https://github.com/commanded/commanded/pull/607
	Update local_cluster by @drteeth in https://github.com/commanded/commanded/pull/610


  
    
  
  v1.4.6



  
    
  
  Enhancements


	Includes changelog updates
	Version bump


  
    
  
  v1.4.5



  
    
  
  Enhancements


	Support OTP 26 and Elixir 1.17 (#595).


  
    
  
  v1.4.4



  
    
  
  Enhancements


	feat: put aggregate_state into assigns of the pipeline (#502).
	Add tag to partition test case (#525).
	Make before_reset/0 an explicit callback function (#550).
	New Event.Handler.after_start/1 callback allows configuration in the handler's process (#568).


  
    
  
  Bug fixes


	Fix EventData typespec (#495).
	Fix refute_receive_event examples (#557).
	Fix interested? function doc (#562).
	Use TypeProvider for process managers snapshot serialization (#558).
	fix(router.ex): Telemetry is not emitted if dispatch fails for {:error, :unregistered_command} (#563).


  
    
  
  v1.4.3



  
    
  
  Enhancements


	Use Logger.warning to fix deprecation warnings (#542).
	Add typespec to CompositeRouter.dispatch/2 function (#536).
	Support opts in Commanded.EventStore.append_to_stream function (#528).
	Process manager metadata access (#514).


  
    
  
  Bug fixes


	 Correct parameter type in ProcessManager.after_command/1 callback function (#533).


  
    
  
  v1.4.2


	Record aggregate state while processing Commanded.Aggregate.Multi (#507).
	Properly handle EXIT signal in event handler (#512).
	Separate logging a process managers error (#513).


  
    
  
  v1.4.1



  
    
  
  Enhancements


	Retry command execution when the aggregate process is down (#494).


  
    
  
  Bug fixes


	Remove duplicate apply function call when receiving missed events published to an aggregate's event stream (364c877).
	Fix typespec typo in Commanded.Application (#503).


  
    
  
  v1.4.0



  
    
  
  Enhancements


	Allow a process manager to stop after dispatching a command (#460).
	Replace use Mix.Config with import Config in config files (#467).
	Event handler concurrency (#486).
	Remove elixir_uuid dependency (#493).
	Support and test for OTP 25 (#489).



  
    
  
  v1.3.1



  
    
  
  Bug fixes


	Event Handler not calling init/1 callback function on restart (#463).
	Call process manager init/1 function on process restart (#464).


  
    
  
  v1.3.0



  
    
  
  Enhancements


	Allow command identity to be provided during dispatch (#406).
	Define Commanded.Telemetry module to emit consistent telemetry events (#414).
	Telemetry [:commanded, :aggregate, :execute] events (#407).
	Telemetry [:commanded, :event, :handle] events (#408).
	Telemetry [:commanded, :process_manager, :handle] events (#418).
	Telemetry [:commanded, :application, :dispatch] (#423).
	Graceful shutdown of event handlers (#431).
	Ensure command dispatch metadata is a map (#432).
	Retry command execution on node down (#429).
	Dispatch returning resultant events (#444).
	Get aggregate state (#448).
	Support telemetry v1.0 (#456).


  
    
  
  v1.2.0



  
    
  
  Enhancements


	Add init/1 callback function to event handlers and process managers (#393).
	Include application and handler_name as additional event handler metadata (#396).
	Allow GenServer start options to be provided when starting event handlers and process managers (#398).
	Add hibernate_after option to application config (#399).
	Add support for providing adapter-specific event store subscription options (#391).
	Support custom state for event handlers (#400).
	Allow event handlers and process manager error callback to return failure context struct (#397).
	Allow a before execute function to be defined which is called with the command dispatch execution context and aggregate state before (#402).


  
    
  
  Bug fixes


	Allow process manager error/3 callback to return :skip for failed commands, not just failed events (#362).



  
    
  
  v1.1.1



  
    
  
  Enhancements


	Capture exception on Process Manager apply/2 and call error/3 callback functions (#380).
	Include metadata in upcaster protocol (#389).


  
    
  
  Bug fixes


	Fix Commanded.Aggregate.Multi.execute/2 calls which return Multi struct (#385).


  
    
  
  v1.1.0



  
    
  
  Enhancements


	Dynamic Commanded applications (#324).
	Log and ignore unexpected messages received by event handlers and process manager instances (#333)
	Process manager identity/0 function (#334).
	Extend Commanded.AggregateCase ExUnit case template to support Commanded.Aggregate.Multi.
	Allow Commanded.Aggregate.Multi to return events as :ok tagged tuples.
	Run the formatter in CI (#341).
	Add stacktraces to EventHandler error logging (#340)
	refute_receive_event/4 only tests newly created events (#347).
	Allow Commanded Application name to be set dynamically in middleware (#352).
	Remove router module compile-time checking (#363).
	Reduce memory consumption during aggregate state rebuild (#368).
	Upgrade to phoenix_pubsub to 2.0 (#365).
	Ignore :not_found error when resetting InMemory event store (#354).
	Add router/1 to locals_without_parens in Mix format config (#351).
	Include stacktrace in event handler and process manager error callback functions (#342).
	Call event handler's error/3 callback function when handle/2 function returns an invalid value (#372).


  
    
  
  Bug fixes


	Fixes the typespec for command dispatch (#325).
	Process manager stops if interested?/1 returns an empty list (#335).



  
    
  
  v1.0.1



  
    
  
  Enhancements


	Global registry using Erlang's :global module (#344).
	Command dispatch return (#331).


  
    
  
  Bug fixes


	Fix distributed subscription registration bug (#345).
	Retry event handler and process manager subscriptions on error (#348).


  
    
  
  v1.0.0



  
    
  
  Breaking changes


	Support multiple Commanded apps (#298).


  
    
  
  Enhancements


	Define adapter behaviour modules for event store, pubsub, and registry (#311).
	Add AggregateCase ExUnit case template to support aggregate unit testing (#315).
	Application config lookup (#318).


  
    
  
  Bug fixes


	Fix process manager exception on start (#307).
	Fix commanded aggregate race (#308).
	Fix Dialyzer warnings and include in Travis CI (#317).


  
    
  
  Upgrading


Follow the upgrade guide to define and use your own Commanded application.


  
    
  
  v0.19.1



  
    
  
  Enhancements


	Reset event handler mix task mix commanded.reset MyApp.Handler (#293).


  
    
  
  Bug fixes


	Fix regression in Commanded.Middleware.Logger.delta (#295).


  
    
  
  v0.19.0



  
    
  
  Enhancements


	Update typespec for data and metadata fields in Commanded.EventStore.EventData struct (#246).
	Add include_execution_result and aggregate_version to typespec for router command dispatch (#262).
	Add .formatter.exs to Hex package (#247).
	Event upcasting (#263).
	Support :ok tagged tuple events from aggregate (#268).
	Modify Commanded.Registration.start_child to pass a child_spec (#273).
	Add supervisor_child_spec/2 to Commanded.Registration behaviour (#277) used by Commanded Horde Registry.
	Ensure Commanded can be compiled when optional Jason dependency is not present (#286).
	Fix Aggregate initialization races (#287).
	Support {:system, varname} format in Phoenix PubSub config (#291).


  
    
  
  Breaking changes


	Use DateTime instead of NaiveDateTime for all datetimes (#254).
  This affects the created_at field defined in the Commanded.EventStore.RecordedEvent. You will need to migrate from NaiveDateTime to DateTime if you use this field in your code (such as in an event handler's metadata).




  
    
  
  v0.18.1



  
    
  
  Enhancements


	Process manager idle process timeout (#290).
	Register event handler and process manager subscriptions on process start (#272).


  
    
  
  v0.18.0



  
    
  
  Enhancements


	Rename uuid dependency to elixir_uuid (#178).
	Allow aggregate identity to be of any type that implements the String.Chars protocol (#166).
	Process manager and event handler error & exception handling (#192).
	Process manager event handling timeout (#193).
	Allow event handlers to subscribe to individual streams (#203).
	Add new values for expected_version for event store append events behaviour (#127).
	Export Commanded.Commands.Router macros in .formatter.exs file (#204).
	Generate specs and docs for Router dispatch functions only once (#206).
	Allow two-arity predicate function in wait_for_event receiving both event data and recorded event struct (#213).
	Allow :infinity timeout on command dispatch (#227)
	Strict process manager routing (#243).
	Allow Commanded.Aggregate.Multi to be nested (#244).
	Add child_spec/0 function to Commanded.EventStore behaviour.
	Add delete_subscription/2 to Commanded.EventStore behaviour (#245).
	Add refute_receive_event/2 to Commanded.Assertions.EventAssertions test helpers.


  
    
  
  Bug fixes


	Fix typo in include_execution_result global router option (#216).
	Handle the {:ok, _} tuple dispatch result in process manager command dispatch (#236).
	Allow string keys for Commanded.Middleware.Pipeline.assign_metadata/3, atoms are being deprecated (#228)
	Fix Commanded.PubSub.subscribe typespec (#222).


  
    
  
  Breaking changes


	Migrate to Jason for JSON serialization (#234).
You will need to add Jason as a dependency in mix.exs:
defp deps do
  [{:jason, "~> 1.2"}]
end
Jason has no support for encoding arbitrary structs - explicit implementation of the Jason.Encoder protocol is always required. You must update all your domain event modules, aggregate state (when using state snapshotting), and process manager state to include @derive Jason.Encoder as shown below:
defmodule AnEvent do
  @derive Jason.Encoder
  defstruct [:field]
end

	Extend aggregate lifespan behaviour to include after_error/1 and after_command/1 callbacks (#210).
Previously you only had to define an after_event/1 callback function to implement the Commanded.Aggregates.AggregateLifespan behaviour:
defmodule BankAccountLifespan do
  @behaviour Commanded.Aggregates.AggregateLifespan

  def after_event(%BankAccountClosed{}), do: :stop
  def after_event(_event), do: :infinity
end
Now you must also define after_command/1 and after_error/1 callback functions:
defmodule BankAccountLifespan do
  @behaviour Commanded.Aggregates.AggregateLifespan

  def after_event(%BankAccountClosed{}), do: :stop
  def after_event(_event), do: :infinity

  def after_command(%CloseAccount{}), do: :stop
  def after_command(_command), do: :infinity

  def after_error(:invalid_initial_balance), do: :stop
  def after_error(_error), do: :stop
end



  
    
  
  Upgrading


Please ensure you upgrade the following event store dependencies.
Using the Elixir EventStore:
	eventstore to v0.16.0
	commanded_eventstore_adapter to v0.5.0

Using EventStoreDB:
	commanded_extreme_adapter to v0.6.0

Commanded Ecto projections:
	commanded_ecto_projections to v0.8.0

Commanded scheduler:
	commanded_scheduler to v0.2.0



  
    
  
  v0.17.5



  
    
  
  Enhancements


	Process manager idle process timeout (#290).


  
    
  
  v0.17.4



  
    
  
  Bug fixes


	Register event handler and process manager subscriptions on process start (#272).


  
    
  
  v0.17.3



  
    
  
  Bug fixes


	Fix snapshot recording (#196).
	Fixed warning about deprecated time unit in elixir 1.8 (#239).


  
    
  
  v0.17.2



  
    
  
  Enhancements


	Remove default error/4 callback function from process manager to silence deprecation warning.


  
    
  
  v0.17.1



  
    
  
  Enhancements


	Support Phoenix.PubSub v1.1.0.


  
    
  
  Bug fixes


	Set default aggregate lifespan timeout to :infinity (#200).


  
    
  
  v0.17.0



  
    
  
  Enhancements


	Ability to globally override include_execution_result and include_aggregate_version in environment config (#168).
	Handle custom type serialization in snapshot source type (#165).
	Fix compiler warnings in generated code (routers, event handlers, and process managers).
	Add InMemory.reset!/0 for testing purposes (#175).


  
    
  
  Bug fixes


	Ensure process managers can be configured with :strong consistency.
	Fix error when subscription process already tracked (#180).



  
    
  
  v0.16.0


	Support composite command routers (#111).
	Aggregate state snapshots (#121).
	New error/3 callback for process manager and deprecated error/4 (#124)
	Router support for identity prefix function.
	Retry command execution on concurrency error (#132).
	Event handler error/3 callback (#133).
	Support distributed dispatch consistency (#135).
	Defer event handler and process router init until after subscribed (#138).
	Replace aggregate lifespan after_command/1 callback with after_event/1 (#139).
	Support process manager routing to multiple instances (#141).
	Allow a default consistency to be set via the application env (#150).
	Command dispatch consistency using explicit handler names (#161).


  
    
  
  Breaking changes


	The Commanded.Aggregates.AggregateLifespan behaviour has been changed from after_command/1 to after_event/1. You will need to update your own lifespan modules to use events instead of commands to shutdown an aggregate process after an inactivity timeout.


  
    
  
  Upgrading


Please ensure you upgrade the following event store dependencies.
Using the Elixir EventStore:
	eventstore to v0.14.0
	commanded_eventstore_adapter to v0.4.0

Using EventStoreDB:
	commanded_extreme_adapter to v0.5.0



  
    
  
  v0.15.1



  
    
  
  Bug fixes


	Event handler child_spec/1 must include config options defined by use macro.


  
    
  
  v0.15.0



  
    
  
  Enhancements


	Process manager command dispatch error handling (#93).
	Event handlers may define an init/0 callback function to start any related processes. It must return :ok, otherwise the handler process will be stopped.
	Add include_execution_result option to command dispatch (#96).
	Add Commanded.Aggregate.Multi (#98) as a way to return multiple events from a command dispatch that require aggregate state to be updated after each event.
	Correlation and causation ids (#105).
	Initial support for running on a cluster of nodes (#80).


  
    
  
  Bug fixes


	Adding a prefix to the aggregate in the router breaks the strong consistency of command dispatch (#101).


  
    
  
  Upgrading


Please ensure you upgrade the following event store dependencies.
Using the Elixir EventStore:
	eventstore to v0.13.0
	commanded_eventstore_adapter to v0.3.0

Using EventStoreDB:
	commanded_extreme_adapter to v0.4.0



  
    
  
  v0.14.0



  
    
  
  Enhancements


	Dispatch command with :eventual or :strong consistency guarantee (#82).
	Additional stream prefix per aggregate (#77).
	Include custom metadata during command dispatch (#61).
	Validate command dispatch registration in router (59).


  
    
  
  Upgrading


Please ensure you upgrade the following event store dependencies.
Using the Elixir EventStore:
	eventstore to v0.11.0
	commanded_eventstore_adapter to v0.2.0

Using EventStoreDB:
	commanded_extreme_adapter to v0.3.0



  
    
  
  v0.13.0



  
    
  
  Enhancements


	Command dispatch optionally returns aggregate version, using include_aggregate_version: true during dispatch.



  
    
  
  v0.12.0



  
    
  
  Enhancements


	Commanded.Event.Handler and Commanded.ProcessManagers.ProcessManager macros to simplify defining, and starting, event handlers and process managers. Note the previous approach to defining and starting may still be used, so this is not a breaking change.



  
    
  
  v0.11.0



  
    
  
  Enhancements


	Shutdown idle aggregate processes (#43).



  
    
  
  v0.10.0



  
    
  
  Enhancements


	Extract event store integration to a behaviour (Commanded.EventStore). This defines the contract to be implemented by an event store adapter. It allows additional event store databases to be used with Commanded.
By default, a GenServer in-memory event store adapter is used. This should only be used for testing as there is no persistence.
The existing PostgreSQL-based eventstore integration has been extracted as a separate package (commanded_eventstore_adapter). There is also a new adapter for Greg Young's Event Store using the Extreme library (commanded_extreme_adapter).
You must install the required event store adapter package and update your environment configuration to specify the :event_store_adapter module. See the README for details.




  
    
  
  v0.9.0



  
    
  
  Enhancements


	Stream events from event store when rebuilding aggregate state.



  
    
  
  v0.8.5



  
    
  
  Enhancements


	Upgrade to Elixir 1.4 and remove compiler warnings.


  
    
  
  v0.8.4



  
    
  
  Enhancements


	Event handler and process manager subscriptions should be created from a given stream position (#14).
	Stop process manager instance after reaching its final state (#24).


  
    
  
  v0.8.3



  
    
  
  Enhancements


	Middleware after_failure callback is executed even when a middleware halts execution.


  
    
  
  v0.8.2



  
    
  
  Bug fixes


	JsonSerializer should ensure event type atom exists when deserializing (#28).


  
    
  
  v0.8.1



  
    
  
  Enhancements


	Command handlers should be optional by default (#30).


  
    
  
  v0.8.0



  
    
  
  Enhancements


	Simplify aggregates and process managers (#31).



  
    
  
  v0.7.1



  
    
  
  Bug fixes


	Restarting aggregate process should load all events from its stream in batches. The Event Store read stream default limit is 1,000 events.


  
    
  
  v0.7.0



  
    
  
  Enhancements


	Command handling middleware allows a command router to define middleware modules that are executed before, and after success or failure of each command dispatch (#12).



  
    
  
  v0.6.3



  
    
  
  Enhancements


	Process manager instance processes event non-blocking to prevent timeout during event processing and any command dispatching. It persists last seen event id to ensure events are handled only once.


  
    
  
  v0.6.2



  
    
  
  Enhancements


	Command dispatch timeout. Allow a timeout value to be configured during command registration or dispatch. This overrides the default timeout of 5 seconds. The same as the default GenServer call timeout.


  
    
  
  Bug fixes


	Fix pending aggregates restarts: supervisor restarts aggregate process but it cannot accept commands (#22).


  
    
  
  v0.6.1



  
    
  
  Enhancements


	Upgrade eventstore mix dependency to v0.6.0 to use support for recorded events created_at as NaiveDateTime.


  
    
  
  v0.6.0



  
    
  
  Enhancements


	Confirm receipt of events in event handler and process manager router (#19).
	Convert keys to atoms when decoding JSON using Poison decoder.
	Prefix process manager instance snapshot uuid with process manager name.
	Multi command dispatch registration in router (#16).



  
    
  
  v0.5.0



  
    
  
  Enhancements


	Include event metadata as second argument to event handlers. An event handler must now implement the Commanded.Event.Handler behaviour consisting of a single handle_event/2 function.



  
    
  
  v0.4.0



  
    
  
  Enhancements


	Macro to assist with building process managers (README).



  
    
  
  v0.3.1



  
    
  
  Enhancements


	Include unit test event assertion function: assert_receive_event/2 (#13).
	Include top level application in mix config.


  
    
  
  v0.3.0



  
    
  
  Enhancements


	Don't persist an aggregate's pending events when executing a command returns an error (#10).


  
    
  
  Bug fixes


	Ensure an aggregate's pending events are persisted in the order they were applied.



  
    
  
  v0.2.1



  
    
  
  Enhancements


	Support integer, atom or strings as an aggregate UUID (#7).


  
    
  
  v0.1.0


Initial release.


  

    
In memory event store
    

Commanded provides an in-memory event store implementation, for test use only, in the module Commanded.EventStore.Adapters.InMemory. This is a transient event store without persistence.

  
    
  
  Configuration


Ensure you configure the in-memory adapter in your environment config file, config/test.exs:
config :my_app, MyApp.App,
  event_store: [
    adapter: Commanded.EventStore.Adapters.InMemory,
    serializer: Commanded.Serialization.JsonSerializer
  ]
You may replace or omit the serializer configuration. By including it here we ensure events used by the tests can be successfully serialized and deserialized.

  
    
  
  Usage


ExUnit case template
You can use ExUnit's case template feature to restart the in-memory event store for each test run.
defmodule InMemoryEventStoreCase do
  use ExUnit.CaseTemplate

  alias Commanded.EventStore.Adapters.InMemory

  setup do
    {:ok, _apps} = Application.ensure_all_started(:my_app)

    on_exit(fn ->
      :ok = Application.stop(:my_app)
    end)
  end
end
Replace both occurrences of :my_app in the above ExUnit case template with the name of your own application.
The reason why your app must be stopped and then started between resetting the event store is to ensure all application processes are restarted with their initial state to prevent state from one test affecting another.
Use the InMemoryEventStoreCase module within any test files that need to use the event store.
defmodule ExampleTest do
  use InMemoryEventStoreCase

  # Define your tests here ...
end
Running your tests
Run mix test as usual to execute the tests using the in-memory event store.


  

    
Getting started
    

Commanded can be installed from the package manager hex as follows.
	Add commanded to your list of dependencies in mix.exs:
 def deps do
   [{:commanded, "~> 1.4"}]
 end

	Optionally add jason to support JSON serialization via Commanded.Serialization.JsonSerializer:
 def deps do
   [{:jason, "~> 1.3"}]
 end

	Fetch mix dependencies:
 $ mix deps.get


	Define a Commanded application module for your app, see the Application guide for details.
 defmodule MyApp.Application do
   use Commanded.Application, otp_app: :my_app
 end

	Configure one of the supported event stores by following the Choosing an Event Store guide.




  

    
Choosing an event store
    

You must decide which event store to use with Commanded. You have a choice between two existing event stores:
	PostgreSQL-based Elixir EventStore (adapter).

	EventStoreDB (adapter).


There is also an in-memory event store adapter for test use only.
Want to use a different event store? Then you will need to write your own event store provider as described below.


  
    
  
  PostgreSQL-based Elixir EventStore


Use :commanded_eventstore_adapter to persist events to a PostgreSQL database. As the name implies, this is the adapter for EventStore, which is open-source event store using PostgreSQL for persistence and implemented in Elixir.


  
    
  
  EventStoreDB


Use :commanded_extreme_adapter to persist events to EventStoreDB: an open-source database, the best data storage solution for event-sourced systems. It can be run as a cluster of nodes containing the same data, which remains available for writes provided at least half the nodes are alive and connected.
The quickest way to get started with EventStoreDB is by using their official EventStoreDB Docker container.
The Commanded adapter uses the :extreme Elixir TCP client to connect to EventStoreDB.

  
    
  
  Running EventStoreDB


You must run EventStoreDB with all projections enabled and standard projections started.
Use the --run-projections=all --start-standard-projections=true flags when running the EventStoreDB executable.


  
    
  
  Writing your own event store provider


To use an alternative event store with Commanded you will need to implement the Commanded.EventStore.Adapter behaviour. This defines the contract to be implemented by an adapter module to allow an event store to be used with Commanded. Tests to verify an adapter conforms to the behaviour are provided in test/event_store_adapter.
You can use one of the existing adapters (commanded_eventstore_adapter or commanded_extreme_adapter) to understand what is required.


  

    
Using Commanded
    

Commanded provides the building blocks for you to create your own Elixir applications following the CQRS/ES pattern.
A separate guide is provided for each of the components you can build:
	Application
	Aggregates
	Commands, registration and dispatch
	Events and handlers
	Process managers

Commanded uses strong consistency for command dispatch (write model) and eventual consistency, by default, for the read model. Receiving an :ok reply from dispatch indicates the command was successfully handled and any created domain events fully persisted to your chosen event store. You may opt into strong consistency for individual event handlers and command dispatch as required.

  
    
  
  Quick overview


Here's an example bank account opening feature built using Commanded to demonstrate its usage.
	Define an OpenBankAccount command:
 defmodule OpenBankAccount do
   defstruct [:account_number, :initial_balance]
 end

	Define a corresponding BankAccountOpened domain event:
 defmodule BankAccountOpened do
   @derive Jason.Encoder
   defstruct [:account_number, :initial_balance]
 end

	Build a BankAccount aggregate to handle the command, protect its business invariants, and return a domain event when successfully handled:
 defmodule BankAccount do
   defstruct [:account_number, :balance]

   # Public command API

   def execute(%BankAccount{account_number: nil}, %OpenBankAccount{account_number: account_number, initial_balance: initial_balance})
     when initial_balance > 0
   do
     %BankAccountOpened{account_number: account_number, initial_balance: initial_balance}
   end

   # Ensure initial balance is never zero or negative
   def execute(%BankAccount{}, %OpenBankAccount{initial_balance: initial_balance})
     when initial_balance <= 0
   do
     {:error, :initial_balance_must_be_above_zero}
   end

   # Ensure account has not already been opened
   def execute(%BankAccount{}, %OpenBankAccount{}) do
     {:error, :account_already_opened}
   end

   # State mutators

   def apply(%BankAccount{} = account, %BankAccountOpened{} = event) do
     %BankAccountOpened{account_number: account_number, initial_balance: initial_balance} = event

     %BankAccount{account |
       account_number: account_number,
       balance: initial_balance
     }
   end
 end

	Define a router module to route the open account command to the bank account aggregate:
 defmodule BankRouter do
   use Commanded.Commands.Router

   dispatch OpenBankAccount, to: BankAccount, identity: :account_number
 end

	Define an application to host the aggregate and supporting processes:
 defmodule BankApp do
   use Commanded.Application,
     otp_app: :bank,
     event_store: [adapter: Commanded.EventStore.Adapters.InMemory]

   router BankRouter
 end
 This application is configured to use in-memory event store included with Commanded for testing.

	Create an event handler module that updates a bank account balance:
 defmodule AccountBalanceHandler do
   use Commanded.Event.Handler,
     application: BankApp,
     name: __MODULE__

   def after_start(_state) do
     with {:ok, _pid} <- Agent.start_link(fn -> 0 end, name: __MODULE__) do
       :ok
     end
   end

   def handle(%BankAccountOpened{initial_balance: initial_balance}, _metadata) do
     Agent.update(__MODULE__, fn _ -> initial_balance end)
   end

   def current_balance do
     Agent.get(__MODULE__, fn balance -> balance end)
   end
 end

	Start the application and event handler processes:
 {:ok, _pid} = BankApp.start_link()
 {:ok, _pid} = AccountBalanceHandler.start_link()
 In a real application you would use a supervisor to start these processes.


Finally, we can dispatch a command to open a new bank account:
:ok = BankApp.dispatch(%OpenBankAccount{account_number: "ACC123456", initial_balance: 1_000})


  

    
Application
    

Commanded allows you to define, supervise, and start your own application module. To use Commanded you must create at least one application. You can create multiple Commanded applications which will run independently, each using its own separately configured event store.
The application expects at least an :otp_app option to be specified. It should point to an OTP application containing the application's configuration.
For example, the application:
defmodule MyApp.Application do
  use Commanded.Application, otp_app: :my_app

  router MyApp.Router
end
Could be configured with:
# config/config.exs
config :my_app, MyApp.Application,
  event_store: [
    adapter: Commanded.EventStore.Adapters.EventStore,
    event_store: MyApp.EventStore
  ],
  pubsub: :local,
  registry: :local
Alternatively, you can include the event store, pubsub, and registry config when defining the application:
defmodule MyApp.Application do
  use Commanded.Application,
    otp_app: :my_app,
    event_store: [
      adapter: Commanded.EventStore.Adapters.EventStore,
      event_store: MyApp.EventStore
    ],
    pubsub: :local,
    registry: :local

  router(MyApp.Router)
end
Finally, you can provide an optional init/1 function to provide runtime configuration.
defmodule MyApp.Application do
  use Commanded.Application, otp_app: :my_app

  def init(config) do
    {:ok, config}
  end
end

  
    
  
  Routing commands


A Commanded application is also a composite router. This provides the router macro allowing you to include one or more router modules for command dispatch.
defmodule MyApp.Application do
  use Commanded.Application, otp_app: :my_app

  router(MyApp.Billing.Router)
  router(MyApp.Customers.Router)
  router(MyApp.Notifications.Router)
end
Once you have defined a router you can dispatch a command using the application module:
:ok = MyApp.Application.dispatch(%RegisterCustomer{id: Commanded.UUID.uuid4(), name: "Ben"})


  

    
Aggregates
    

Build your aggregates using standard Elixir modules and functions, with structs to hold state. There is no external dependency requirement.
An aggregate is comprised of its state, public command functions, and state mutators.
In a CQRS application all state must be derived from the published domain events. This prevents tight coupling between aggregate instances, such as querying for their state, and ensures their state isn't exposed.
If an aggregate needs data owned by another aggregate, then you should lookup the data from a projection and include it in the command before dispatch. Usually this would be a projection into a read model built for querying. You can use Commanded Ecto projections to project events into a SQL database.

  
    
  
  Aggregate state


Use the defstruct keyword to define the aggregate state and fields.
defmodule ExampleAggregate do
  defstruct [:uuid, :name]
end

  
    
  
  Command functions


A command function receives the aggregate's state and the command to execute. It must return the resultant domain events, which may be one event or multiple events. You can return a single event or a list of events: %Event{}, [%Event{}], {:ok, %Event{}}, or {:ok, [%Event{}]}.
To respond without returning an event you can return :ok, nil or an empty list as either [] or {:ok, []}.
For business rule violations and errors you may return an {:error, error} tagged tuple or raise an exception.
Name your public command functions execute/2 to dispatch commands directly to the aggregate without requiring an intermediate command handler.
defmodule ExampleAggregate do
  def execute(%ExampleAggregate{uuid: nil}, %Create{} = command) do
    %Create{uuid: uuid, name: name} = command

    event = %Created{uuid: uuid, name: name}

    {:ok, event}
  end

  def execute(%ExampleAggregate{}, %Create{}),
    do: {:error, :already_created}
end

  
    
  
  State mutators


The state of an aggregate can only be mutated by applying a domain event to its state. This is achieved by an apply/2 function that receives the state and the domain event. It returns the modified state.
Pattern matching is used to invoke the respective apply/2 function for an event. These functions MUST NOT fail as they are used when rebuilding the aggregate state from its history of domain events. You cannot reject the event once it has occurred.
defmodule ExampleAggregate do
  def apply(%ExampleAggregate{}, %Created{} = event) do
    %Created{uuid: uuid, name: name} = event

    %ExampleAggregate{uuid: uuid, name: name}
  end
end

  
    
  
  Example aggregate


You can define your aggregate with public API functions using the language of your domain. In this bank account example, the public function to open a new account is open_account/3:
defmodule BankAccount do
  defstruct [:account_number, :balance]

  # Public API

  def open_account(%BankAccount{account_number: nil}, account_number, initial_balance)
    when initial_balance > 0
  do
    event = %BankAccountOpened{account_number: account_number, initial_balance: initial_balance}

    {:ok, event}
  end

  def open_account(%BankAccount{}, _account_number, initial_balance)
    when initial_balance <= 0
  do
    {:error, :initial_balance_must_be_above_zero}
  end

  def open_account(%BankAccount{}, _account_number, _initial_balance) do
    {:error, :account_already_opened}
  end

  # State mutators

  def apply(%BankAccount{} = account, %BankAccountOpened{} = event) do
    %BankAccountOpened{account_number: account_number, initial_balance: initial_balance} = event

    %BankAccount{account |
      account_number: account_number,
      balance: initial_balance
    }
  end
end
With this approach you must dispatch the command to a command handler:
defmodule Commanded.ExampleDomain.OpenAccountHandler do
  alias BankAccount
  alias Commands.OpenAccount

  @behaviour Commanded.Commands.Handler

  def handle(%BankAccount{} = aggregate, %OpenAccount{} = command) do
    %OpenAccount{account_number: account_number, initial_balance: initial_balance} = command

    BankAccount.open_account(aggregate, account_number, initial_balance)
  end
end
An alternative approach is to expose one or more public command functions, execute/2, and use pattern matching on the command argument. With this approach you can route your commands directly to the aggregate.
In this example the execute/2 function pattern matches on the OpenAccount command module:
defmodule BankAccount do
  defstruct [:account_number, :balance]

  # Public API

  def execute(
    %BankAccount{account_number: nil},
    %OpenAccount{initial_balance: initial_balance} = command
    ) when initial_balance > 0
  do
    %OpenAccount{account_number: account_number} = command

    %BankAccountOpened{account_number: account_number, initial_balance: initial_balance}
  end

  def execute(%BankAccount{}, %OpenAccount{initial_balance: initial_balance})
    when initial_balance <= 0
  do
    {:error, :initial_balance_must_be_above_zero}
  end

  def execute(%BankAccount{}, %OpenAccount{}) do
    {:error, :account_already_opened}
  end

  # State mutators

  def apply(%BankAccount{} = account, %BankAccountOpened{} = event) do
    %BankAccountOpened{account_number: account_number, initial_balance: initial_balance} = event

    %BankAccount{account |
      account_number: account_number,
      balance: initial_balance
    }
  end
end
Effective Aggregate State
The aggregate state should be carefully designed to maintain only the data required to:
	Enforce business rules and invariants (e.g. preventing withdrawals that would overdraw an account)
	Provide context for command handling based on previous events (e.g. checking an order's current status)
	Make decisions that depend on historical events (e.g. verifying a refund doesn't exceed original payment)

A common anti-pattern is blindly copying all event data into aggregate state without considering whether that data
is actually needed for command handling. If a piece of state is never referenced in any command handling logic,
it should be removed from the aggregate. This keeps the aggregate focused and maintainable.
State that is only needed for querying/reporting should be maintained in read models rather than aggregate state.

  
    
  
  Using Commanded.Aggregate.Multi to return multiple events


Sometimes you need to create multiple events from a single command. You can use Commanded.Aggregate.Multi to help track the events and update the aggregate state. This can be useful when you want to emit multiple events that depend upon the aggregate state being updated.
Any errors encountered will be returned to the caller, the modified aggregate state and any pending events are discarded.

  
    
  
  Example


In the example below, money is withdrawn from the bank account and the updated balance is used to check whether the account is overdrawn.
defmodule BankAccount do
  defstruct [
    account_number: nil,
    balance: 0,
    state: nil,
  ]

  alias Commanded.Aggregate.Multi

  # Public API

  def execute(
    %BankAccount{state: :active} = account,
    %WithdrawMoney{amount: amount})
    when is_number(amount) and amount > 0
  do
    account
    |> Multi.new()
    |> Multi.execute(&withdraw_money(&1, amount))
    |> Multi.execute(&check_balance/1)
  end

  # State mutators

  def apply(%BankAccount{} = state, %MoneyWithdrawn{balance: balance}),
    do: %BankAccount{state | balance: balance}

  def apply(%BankAccount{} = state, %AccountOverdrawn{}),
    do: %BankAccount{state | state: :overdrawn}

  # Private helpers

  defp withdraw_money(%BankAccount{} = account, amount) do
    %BankAccount{account_number: account_number, balance: balance} = account

    event = %MoneyWithdrawn{
      account_number: account_number,
      amount: amount,
      balance: balance - amount
    }

    {:ok, event}
  end

  defp check_balance(%BankAccount{balance: balance} = account)
    when balance < 0
  do
    %BankAccount{account_number: account_number} = account

    event = %AccountOverdrawn{account_number: account_number, balance: balance}

    {:ok, event}
  end

  defp check_balance(%BankAccount{}), do: :ok
end

  
    
  
  Aggregate state snapshots


A snapshot represents the aggregate state when all events to that point in time have been replayed. You can optionally configure state snapshotting for individual aggregates in your app configuration. Instead of loading every event for an aggregate when rebuilding its state, only the snapshot and any events appended since its creation are read. By default snapshotting is disabled for all aggregates.
As an example, assume a snapshot was taken after persisting an event for the aggregate at version 100. When the aggregate process is restarted we load and deserialize the snapshot data as the aggregate's initial state. Then we fetch and replay the aggregate's events after version 100.
This is a performance optimisation for aggregates that have a long lifetime or raise a large number of events. It limits the worst case scenario when rebuilding the aggregate state: it will need to read the snapshot and at most this many events from storage.
Use the following options to configure snapshots for an aggregate:
	snapshot_every - snapshot aggregate state every so many events. Use nil to disable snapshotting, or exclude the configuration entirely.

	snapshot_version - a non-negative integer indicating the version of the aggregate state snapshot. Incrementing this version forces any earlier recorded snapshots to be ignored when rebuilding aggregate state.



  
    
  
  Example


In config/config.exs enable snapshots for ExampleAggregate after every ten events:
config :my_app, MyApp.Application,
  snapshotting: %{
    MyApp.ExampleAggregate => [
      snapshot_every: 10,
      snapshot_version: 1
    ]
  }

  
    
  
  Snapshot serialization


Aggregate state will be serialized using the configured event store serializer, by default this stores the data as JSON. Remember to derive the Jason.Encoder protocol for the aggregate state to ensure JSON serialization is supported, as shown below.
defmodule ExampleAggregate do
  @derive Jason.Encoder
  defstruct [:name, :datetime]
end
You can use the Commanded.Serialization.JsonDecoder protocol to decode the parsed JSON data into the expected types:
defimpl Commanded.Serialization.JsonDecoder, for: ExampleAggregate do
  @doc """
  Parse the datetime included in the aggregate state
  """
  def decode(%ExampleAggregate{} = state) do
    %ExampleAggregate{datetime: datetime} = state

    {:ok, dt, _} = DateTime.from_iso8601(datetime)

    %ExampleAggregate{state | datetime: dt}
  end
end
Note: The default JSON encoding of a DateTime struct uses the to_iso8601/1 function which is why we must decode it using the from_iso8601/1 function.

  
    
  
  Rebuilding an aggregate snapshot


Whenever you change the structure of an aggregate's state you MUST increment the snapshot_version number. The aggregate state will be rebuilt from its events, ignoring any existing snapshots. They will be overwritten when the next snapshot is taken.


  

    
Commands
    

You need to create a module per command and define the fields using defstruct:
defmodule OpenAccount do
  @enforce_keys [:account_number]
  defstruct [:account_number, :initial_balance]
end
A command must contain a field to uniquely identify the aggregate instance (e.g. account_number). Use @enforce_keys to force the identity field to be specified when creating the command struct.
Since commands are just plain Elixir structs you can use a library such as typed_struct for defining structs, fields with their types, and enforcing mandatory keys without writing too much boilerplate code.
Commanded logs commands during execution when using debug log level. To prevent sensitive data from being exposed in
logs you can use the built-in Elixir Inspect module to exclude fields.
defmodule RegisterUser do
  @derive {Inspect, only: [:id, :name]}
  defstruct [:id, :name, :password]
end

  
    
  
  Command dispatch and routing


A router module is used to route commands to their registered command handler and aggregate module.
You create a router module using Commanded.Commands.Router and register each command with its associated handler:
defmodule BankRouter do
  use Commanded.Commands.Router

  dispatch OpenAccount, to: OpenAccountHandler, aggregate: BankAccount, identity: :account_number
  dispatch DepositMoney, to: DepositMoneyHandler, aggregate: BankAccount, identity: :account_number
end
This can be more succinctly configured by excluding command handlers and dispatching directly to the aggregate, using multi-command registration, and with the identify helper macro:
defmodule BankRouter do
  use Commanded.Commands.Router

  identify BankAccount, by: :account_number
  dispatch [OpenAccount, DepositMoney], to: BankAccount
end
A router must be registered with its associated application using the router macro, as shown below:
defmodule BankApp do
  use Commanded.Application, otp_app: :bank_app

  router BankRouter
end

  
    
  
  Command handlers


A command handler receives the aggregate and the command being executed. It allows you to validate, authorize, and/or enrich the command with additional data before executing the appropriate aggregate module function.
The command handler must implement the Commanded.Commands.Handler behaviour consisting of a single handle/2 function. It receives the aggregate state and the command to be handled. It must return the raised domain events from the aggregate. It may return an {:error, reason} tuple on failure. The handle/2 function is executed in the same process as the aggregate itself and is therefore within the consistency boundary of the aggregate. In effect, there is no essential difference between using the handle/2 function in a command handler module vs. the execute/2 function in the aggregate module aside from organization of your source code.
defmodule OpenAccountHandler do
  @behaviour Commanded.Commands.Handler

  def handle(%BankAccount{} = aggregate, %OpenAccount{} = command) do
    %OpenAccount{account_number: account_number, initial_balance: initial_balance} = command

    BankAccount.open_account(aggregate, account_number, initial_balance)
  end
end
Command handlers execute in the context of the dispatch call, as such they are limited to the timeout period specified. The default timeout is five seconds, the same as a GenServer call. You can increase the timeout value for individual commands as required - see the section on Timeouts below.

  
    
  
  Dispatch directly to aggregate


It is also possible to route a command directly to an aggregate, without requiring an intermediate command handler.
defmodule BankRouter do
  use Commanded.Commands.Router

  # will route to `BankAccount.execute/2`
  dispatch OpenAccount, to: BankAccount, identity: :account_number
end
By default, the aggregate module's execute/2 function will be called with the aggregate's state and the command to execute. Using this approach, you will create an execute/2 clause that pattern-matches on each command that the aggregate should handle.
Alternatively, you may specify the name of a function (also receiving both the aggregate state and the command) on your aggregate module to which the command will be dispatched:
defmodule BankRouter do
  use Commanded.Commands.Router

  # Will route to `BankAccount.open_account/2`
  dispatch OpenAccount, to: BankAccount, function: :open_account, identity: :account_number
end

  
    
  
  Dispatching commands


You dispatch a command to its registered aggregate using the application:
:ok = BankApp.dispatch(%OpenAccount{account_number: "ACC123", initial_balance: 1_000})
Optionally, you can dispatch a command using the router by providing the application as an option:
command = %OpenAccount{account_number: "ACC123", initial_balance: 1_000}

:ok = BankRouter.dispatch(command, application: BankApp)

  
    
  
  Define aggregate identity


You can define the identity field for an aggregate once using the identify macro. The configured identity will be used for all commands registered to the aggregate, unless overridden by a command registration.
Example
defmodule BankRouter do
  use Commanded.Commands.Router

  identify BankAccount, by: :account_number

  dispatch OpenAccount, to: BankAccount
end
The above configuration requires that all commands for the BankAccount aggregate must contain an account_number field used to identity a unique instance.
Identity prefix
An optional identity prefix can be used to distinguish between different aggregates that would otherwise share the same identity. As an example you might have a User and a UserPreferences aggregate that you wish to share the same identity. In this scenario you should specify a prefix for each aggregate (e.g. "user-" and "user-preference-").
defmodule BankRouter do
  use Commanded.Commands.Router

  identify BankAccount,
    by: :account_number,
    prefix: "bank-account-"

  dispatch OpenAccount, to: BankAccount
end
The prefix is used as the stream identity when appending, and reading, the aggregate's events (e.g. <prefix><instance_identity>). Note you must not change the stream prefix once you have events persisted in your event store, otherwise the aggregate's events cannot be read from the event store and its state cannot be rebuilt since the stream name will be different.
Custom aggregate identity
Any module that implements the String.Chars protocol can be used for an aggregate's identity. By default this includes the following Elixir built-in types: strings, integers, floats, atoms, and lists.
You can define your own custom identity modules and implement the String.Chars protocol for them:
defmodule AccountNumber do
  defstruct [:branch, :account_number]

  defimpl String.Chars do
    def to_string(%AccountNumber{branch: branch, account_number: account_number}),
      do: branch <> ":" <> account_number
  end
end
The custom identity will be converted to a string during command dispatch. This is used as the aggregate's identity and determines the stream to append its events in the event store.
open_account = %OpenAccount{
  account_number: %AccountNumber{branch: "B1", account_number: "ACC123"},
  initial_balance: 1_000
}

:ok = BankApp.dispatch(open_account)

  
    
  
  Timeouts


A command handler has a default timeout of 5 seconds. The same default as a GenServer.call/3 process call. It must handle the command in this period, otherwise the call fails and the caller process exits.
You can configure a different timeout value during command registration by providing a timeout option, defined in milliseconds:
defmodule BankRouter do
  use Commanded.Commands.Router

  # Configure a timeout of 1 second for the open account command handler
  dispatch OpenAccount,
    to: OpenAccountHandler,
    aggregate: BankAccount,
    identity: :account_number,
    timeout: 1_000
end
You can override the timeout value during command dispatch. This example is dispatching the open account command with a timeout of 2 seconds:
open_account = %OpenAccount{account_number: "ACC123", initial_balance: 1_000}

:ok = BankApp.dispatch(open_account, timeout: 2_000)

  
    
  
  Multi-command registration


Command routers support multi command registration so you can group related command handlers into the same module:
defmodule BankRouter do
  use Commanded.Commands.Router

  dispatch [OpenAccount,CloseAccount],
    to: BankAccountHandler,
    aggregate: BankAccount,
    identity: :account_number
end

  
    
  
  Command dispatch consistency guarantee


You can choose the consistency guarantee when dispatching a command.
	Strong consistency offers up-to-date data but at the cost of high latency.
	Eventual consistency offers low latency but read model queries may reply with stale data since they may not have processed the persisted events.

In Commanded, the available options during command dispatch are:
	:eventual (default) - don't block command dispatch and don't wait for any event handlers, regardless of their own consistency configuration.
:ok = BankApp.dispatch(command)
:ok = BankApp.dispatch(command, consistency: :eventual)

	:strong - block command dispatch until all strongly consistent event handlers and process managers have successfully processed all events created by the command.
:ok = BankApp.dispatch(command, consistency: :strong)
Dispatching a command using :strong consistency but without any strongly consistent event handlers configured will have no effect.

	Provide an explicit list of event handler and process manager modules (or their configured names), containing only those handlers you'd like to wait for. No other handlers will be awaited on, regardless of their own configured consistency setting.
:ok = BankApp.dispatch(command, consistency: [ExampleHandler, AnotherHandler])
:ok = BankApp.dispatch(command, consistency: ["ExampleHandler", "AnotherHandler"])
Note you cannot opt-in to strong consistency for a handler that has been configured as eventually consistent.


Which consistency guarantee should I use?
When dispatching a command using consistency: :strong the dispatch will block until all of the strongly consistent event handlers and process managers have handled all events created by the command. This guarantees that when you receive the :ok response from dispatch, your strongly consistent read models will have been updated and can safely be queried.
Strong consistency helps to alleviate problems and workarounds you would otherwise encounter when dealing with eventual consistency in your own application. Use :strong consistency when you want to query a read model immediately after dispatching a command. You must also configure the event handler to use :strong consistency.
Using :eventual consistency, or omitting the consistency option, will cause the command dispatch to immediately return without waiting for any event handlers or process managers. The handlers run independently, and asynchronously, in the background, therefore you will need to deal with potentially stale read model data.
Configure default consistency
You may override the default consistency (:eventual) by setting default_consistency in your environment config (e.g. config/config.exs):
config :commanded, default_consistency: :strong
This will effect command dispatch, event handlers, and process managers where a consistency is not explicitly defined.
Consistency failures
By opting-in to strong consistency you may encounter an additional error reply from command dispatch:
case BankApp.dispatch(command, consistency: :strong) do
  :ok -> # ... all ok
  {:error, :consistency_timeout} -> # command ok, handlers have not yet executed
end
Receiving an {:error, :consistency_timeout} error indicates the command successfully dispatched, but some or all of the strongly consistent event handlers have not yet executed.
The default timeout is configured as five seconds; this determines how long the dispatch will block waiting for the handlers. You can override the default value in your environment config file (e.g. config/config.exs):
config :commanded,
  dispatch_consistency_timeout: 10_000  # ten seconds

  
    
  
  Dispatch returning execution result


You can choose to include the execution result as part of the dispatch result by setting include_execution_result true:
{
  :ok,
  %Commanded.Commands.ExecutionResult{
    aggregate_uuid: aggregate_uuid,
    aggregate_version: aggregate_version,
    events: events,
    metadata: metadata
  }
} = BankApp.dispatch(command, include_execution_result: true)
This is useful if you need to get information from the events produced by the aggregate.

  
    
  
  Dispatch returning aggregate version


You can optionally choose to include the aggregate's version as part of the dispatch result by setting the include_aggregate_version option to true:
{:ok, aggregate_version} = BankApp.dispatch(command, include_aggregate_version: true)
The returned aggregate_version can be used as an ETAG, allowing you to synchronize operations across the read-side and write-side of your application. For example, if need to wait for an event handler, such as a read model projection, to be up-to-date before continuing execution or querying its data.
defmodule BankAccountProjector do
  use Commanded.Projections.Ecto,
    application: BankApp,
    name: "BankAccountProjector",
    repo: BankApp.Repo

  project(%BankAccountOpened{} = event, metadata, fn multi ->
    multi
    |> Ecto.Multi.run(:bank_account, &find_bank_account(&1, &2, event.id))
    |> Ecto.Multi.update(
      :updated_bank_account,
      &BankAccount.changeset(&1.bank_account, %{
        # Notice that I am using the stream version from the metadata
        # to update the auction's stream version.
        stream_version: metadata.stream_version,
      })
    )
  end)
end
Then you can use the aggregate version to wait for the event handler to be up-to-date:
defmodule BankAccounts do
  def get_bank_account(id, stream_version) do
    query =
      from b in BankAccount,
        where: b.id == ^id,
        where: b.stream_version >= ^stream_version,
        order_by: [asc: b.stream_version]

    case BankApp.one(query) do
      nil ->
        # not ready yet means the projection is not yet up-to-date
        # so you can retry the query
        {:error, :not_ready}

      bank_account ->
        {:ok, bank_account}
    end
  end
end

  
    
  
  Aggregate version as an ETAG


The aggregate version can be thought of as an ETAG (Entity Tag) for a given resource. ETAGs are commonly used in HTTP caching mechanisms to determine if a resource has changed. Similarly, the aggregate version serves as a unique identifier for a specific state of the aggregate.
When building a REST API, you can use the aggregate version as the ETAG in your HTTP responses.

  
    
  
  Causation and correlation ids


To assist with monitoring and debugging your deployed application it is useful to track the causation and correlation ids for your commands and events.
	causation_id - the UUID of the command causing an event, or the event causing a command dispatch.
	correlation_id - a UUID used to correlate related commands/events.

You can set causation and correlation ids when dispatching a command:
:ok = BankApp.dispatch(command, causation_id: Commanded.UUID.uuid4(), correlation_id: Commanded.UUID.uuid4())
When dispatching a command in an event handler, you should copy these values from the metadata (second) argument associated with the event you are handling:
defmodule ExampleHandler do
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: "ExampleHandler"

  def handle(%AnEvent{..}, %{event_id: causation_id, correlation_id: correlation_id}) do
    command = %ExampleCommand{..}

    BankApp.dispatch(command,
      causation_id: causation_id,
      correlation_id: correlation_id,
    )
  end
end
Commands dispatched by a process manager will be automatically assigned the appropriate causation and correlation ids from the source domain event.
You can use Commanded audit middleware to record every dispatched command. This allows you to follow the chain of commands and events by using the causation id. The correlation id can be used to find all related commands and events.

  
    
  
  Event metadata


It's helpful for debugging to have additional metadata associated with events issued by a command. You can set it when dispatching a command:
:ok = BankApp.dispatch(command, metadata: %{"issuer_id" => issuer_id, "user_id" => "user@example.com"})
Note, due to metadata serialization you should expect that only: strings, numbers, and boolean values are preserved; any other value will be converted to a string.
You should always use string keys in your metadata map; atom keys will be converted to strings.

  
    
  
  Aggregate lifespan


By default an aggregate instance process will run indefinitely once started. You can control this by implementing the Commanded.Aggregates.AggregateLifespan behaviour in a module.
Define a module that implements the Commanded.Aggregates.AggregateLifespan behaviour:
defmodule BankAccountLifespan do
  @behaviour Commanded.Aggregates.AggregateLifespan

  def after_event(%MoneyDeposited{}), do: :timer.hours(1)
  def after_event(%BankAccountClosed{}), do: :stop
  def after_event(_event), do: :infinity

  def after_command(%CloseAccount{}), do: :stop
  def after_command(_command), do: :infinity

  def after_error(:invalid_initial_balance), do: :timer.minutes(5)
  def after_error(_error), do: :stop
end
Then specify the module as the lifespan option when registering the command in the router.
defmodule BankRouter do
  use Commanded.Commands.Router

  dispatch [OpenAccount,CloseAccount],
    to: BankAccount,
    lifespan: BankAccountLifespan,
    identity: :account_number
end
The inactivity timeout is specified in milliseconds, after which time the aggregate process will be stopped if no other messages are received by it.
Return :stop or {:stop, reason} to immediately shutdown the aggregate process. Return :infinity to prevent the aggregate instance from shutting down.
You can also return :hibernate and the process is hibernated, it will continue its loop once a message is in its message queue. Hibernating an aggregate causes garbage collection and minimises the memory used by the process. Hibernating should not be used aggressively as too much time could be spent garbage collecting.

  
    
  
  Middleware


Allows a command router to define middleware modules that are executed before and after success or failure of each command dispatch.
This provides an extension point to add in command validation, authorization, logging, and other cross-cutting behaviour that you want to be called for every command the router dispatches.
defmodule BankingRouter do
  use Commanded.Commands.Router

  middleware CommandLogger
  middleware MyCommandValidator
  middleware AuthorizeCommand

  identify BankAccount, by: :account_number

  dispatch OpenAccount, to: OpenAccountHandler, aggregate: BankAccount
  dispatch DepositMoney, to: DepositMoneyHandler, aggregate: BankAccount
end
The middleware modules are executed in the order they’ve been defined. They will receive a Commanded.Middleware.Pipeline struct containing the command being dispatched, in the :command field.

  
    
  
  Example middleware


Implement the Commanded.Middleware behaviour in your module and define the before_dispatch, after_dispatch, and after_failure callback functions.
defmodule NoOpMiddleware do
  @behaviour Commanded.Middleware

  alias Commanded.Middleware.Pipeline
  import Pipeline

  def before_dispatch(%Pipeline{} = pipeline) do
    pipeline
  end

  def after_dispatch(%Pipeline{} = pipeline) do
    pipeline
  end

  def after_failure(%Pipeline{} = pipeline) do
    pipeline
  end
end
Commanded provides a Commanded.Middleware.Logger middleware for logging the name of each dispatched command and its execution duration.

  
    
  
  Composite command routers


You can use the Commanded.Commands.CompositeRouter macro to define a router comprised of other router modules. This approach is useful if you prefer to construct a router per context and then combine them together to form a top level application router.
By using Commanded.Commands.CompositeRouter in your own module you can include other routers via the router macro:
defmodule ApplicationRouter do
  use Commanded.Commands.CompositeRouter

  router BankAccountRouter
  router MoneyTransferRouter
end
defmodule BankApp do
  use Commanded.Application, otp_app: :bank_app

  router ApplicationRouter
end
Command dispatch works the same as any other router:
:ok = BankApp.dispatch(%OpenAccount{account_number: "ACC123", initial_balance: 1_000})


  

    
Events
    


  
    
  
  Domain events


Domain events indicate that something of importance has occurred, within the context of an aggregate. They should be named in the past tense: account registered; funds transferred; fraudulent activity detected etc.
Create a module per domain event and define the fields with defstruct. An event should contain a field to uniquely identify the aggregate instance (e.g. account_number).
Remember to derive the Jason.Encoder protocol for the event struct to ensure JSON serialization is supported, as shown below.
defmodule BankAccountOpened do
  @derive Jason.Encoder
  defstruct [:account_number, :initial_balance]
end
Note, due to event serialization you should expect that only: strings, numbers and boolean values defined in an event are preserved; any other value will be converted to a string. You can control this behaviour as described in the Serialization guide.

  
    
  
  Event handlers


Event handlers allow you to execute code that reacts to domain events: to build read model projections; dispatch commands to other aggregates; and to interact with third-party systems such as sending emails.
Commanded guarantees only one instance of an event handler will run, regardless of how many nodes are running (even when not using distributed Erlang). This is enforced by the event store subscription (PostgreSQL advisory locks in Elixir Event Store).
Use the Commanded.Event.Handler macro within your event handler module to implement the defined behaviour. This consists of a single handle/2 function that receives each published domain event and its metadata, including the event's unique event number. It should return :ok on success or {:error, :reason} on failure. You can return {:error, :already_seen_event} to skip events that have already been handled, due to the at-least-once event delivery of the supported event stores.
Use pattern matching to match on each type of event you are interested in. A catch-all handle/2 function is included, so all other events will be ignored by default.
defmodule ExampleHandler do
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: "ExampleHandler"

  @impl Commanded.Event.Handler
  def handle(%AnEvent{..}, _metadata) do
    # ... process the event
    :ok
  end
end
The name given to the event handler must be unique and remain unchanged between releases. It is used when subscribing to the event store to track which events the handler has seen during restarts.
{:ok, _handler} = ExampleHandler.start_link()

  
    
  
  Subscription options


You can choose to start the event handler's event store subscription from the :origin, :current position, or an exact event number using the start_from option. The default is to use the origin so your handler will receive all events.
defmodule ExampleHandler do
  # Define `start_from` as one of :origin, :current, or an explicit event number (e.g. 1234)
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: "ExampleHandler",
    start_from: :origin
end
You can optionally override :start_from by passing it as param:
{:ok, _handler} = ExampleHandler.start_link(start_from: :current)
Use the :current position when you don't want newly created event handlers to go through all previous events. An example would be adding an event handler to send transactional emails to an already deployed system containing many historical events.
You should start your event handlers using an OTP Supervisor to ensure they are restarted on error. See the Supervision guide for more details.

  
    
  
  Configuration options


You can choose the default error behaviour for all of your event handlers in each Application's configuration:
config :example, ExampleApp,
  on_event_handler_error: :stop # or :backoff or MyCustomErrorHandler
The default behaviour is to stop the event handler process when any error is encountered. As event handlers are supervised either by a custom supervisor or by the application itself, the handlers are usually restarted right away. If the error is permanent, due to a logic or data bug, then the process will likely crash again right away. This can lead the supervisor itself to give up, crash and this will continue up your supervision tree until it stops your application.
The :backoff option, introduced in v1.4.7, cause the even handler to retry after an exponentially increasing backoff period (up to a maximum of 24 hours). The event handler will still not be able to make forward progress until you address the issue, but it won't take your supervisors or applications down with it.
If you want to provide your own strategy, you can pass in a module that implements the Commanded.Event.Handler.error/3 callback.
It's important to note that if your event handler overrides the error/3 callback, then that will be called instead of the application-wide strategy.

  
    
  
  Subscribing to an individual stream


By default event handlers will subscribe to all events appended to any stream. Provide a subscribe_to option to subscribe to a single stream.
defmodule ExampleHandler do
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: __MODULE__,
    subscribe_to: "stream1234"
end
This will ensure the handler only receives events appended to that stream.

  
    
  
  Handling errors


It is important to consider how errors in the handling of events will affect your application. By default any errors encountered are configured to stop the event handler immediately. For this reason it is vital to have handlers under supervision. Errors in handlers occur in two categories:
Transient errors
Transient errors occur due temporary issues such as network connectivity, flakey downstream services, memory, or disk issues. Retrying here will often solve the problem.
Permanent errors
Permanent errors occure due to bugs in code, and as such have no hope of resolving themselves. The handler will not be able to make any progress, and will repeatedly crash.
Repeated crashes become a problem when the handler crashes more than it's supervisor is configured to tolerate. When that happens, the supervisor itself will crash and this process will continue upwards until finally the application itself crashes. This is obviously bad.
You can opt to have your event handlers backoff when crashing to avoid this:
config :my_app, MyApp.CommandedApp,
  on_event_handler_error: :backoff
This will cause all event handlers from MyApp.CommandedApp to back off exponentially when they encounter an error. Errors will backoff to maximum of 24 hours, and will continue at that rate until the issue is resolved. The event handler still can't make any forward progress, but it will at least not bring down your application.
It is also possible to configure the error handler for the default behaviour :stop.
You can bring your own Commanded application-level behaviour by specifiying a module which implements Commanded.Event.Handler.error/3
defmodule YoloErrorHandler do
  def error(error, failing_event, failure_context) do
    # Ignore all errors!
    :skip
  end
end

config :my_app, MyApp.CommandedApp,
  on_event_handler_error: YoloErrorHandler

  
    
  
  Event handler callbacks


	Commanded.Event.Handler.init/1 - (optional) used to configure the handler before it starts.
	Commanded.Event.Handler.after_start/1 - (optional) initialisation callback function called in the process of the started handler.
	Commanded.Event.Handler.error/3 - (optional) called when an event handle/2 callback returns an error.

Error event handlers configured on a per handler basis like this will override the application level error handling.

  
    
  
  Metadata


The handle/2 function in your handler receives the domain event and a map of metadata associated with that event. You can provide the metadata key/value pairs when dispatching a command:
:ok = ExampleApp.dispatch(command, metadata: %{"issuer_id" => issuer_id, "user_id" => "user@example.com"})
In addition to the metadata key/values you provide, the following system values will be included in the metadata passed to an event handler:
	application - the Commanded.Application associated with the event handler.
	handler_name - the name of the event handler.
	state - optional event handler state.
	event_id - a globally unique UUID to identify the event.
	event_number - a globally unique, monotonically incrementing and gapless integer used to order the event amongst all events.
	stream_id - the stream identity for the event.
	stream_version - the version of the stream for the event.
	causation_id - an optional UUID identifier used to identify which command caused the event.
	correlation_id - an optional UUID identifier used to correlate related commands/events.
	created_at - the datetime, in UTC, indicating when the event was created.

These key/value metadata pairs will use atom keys to differentiate them from the user provided metadata:
defmodule ExampleHandler do
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: "ExampleHandler"

  @impl Commanded.Event.Handler
  def handle(event, metadata) do
    IO.inspect(metadata)
    # %{
    #   :causation_id => "db1ebd30-7d3c-40f7-87cd-12cd9966df32",
    #   :correlation_id => "1599630b-9c38-433c-9548-0dd793108ba0",
    #   :created_at => #DateTime<2017-10-30 11:19:56.178901Z>,
    #   :event_id => "5e4a0f38-385b-4d57-823b-a1bcf705b7bb",
    #   :event_number => 12345,
    #   :stream_id => "e42a588d-2cda-4314-a471-5d008cce01fc",
    #   :stream_version => 1,
    #   "issuer_id" => "0768d69a-d2b7-48f4-d0e9-083a97f7ebe0",
    #   "user_id" => "user@example.com"
    # }

    :ok
  end
end

  
    
  
  Consistency guarantee


You can specify an event handler's consistency guarantee using the consistency option:
defmodule ExampleHandler do
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: "ExampleHandler",
    consistency: :eventual
The available options are :eventual (default) and :strong:
	Strong consistency offers up-to-date data but at the cost of high latency.
	Eventual consistency offers low latency but read model queries may reply with stale data since they may not have processed the persisted events.

You request the consistency guarantee, either :strong or :eventual, when dispatching a command. Strong consistency will block the command dispatch and wait for all strongly consistent event handlers to successfully process all events created by the command. Whereas eventual consistency will immediately return after command dispatch, without waiting for any event handlers, even those configured for strong consistency.

  
    
  
  How does it work?


An event handler is a GenServer process that subscribes to the configured event store. For each event persisted to the store the handle/2 callback is called, passing the domain event and its metadata.

  
    
  
  Upcasting events


Commanded supports upcasting of events at runtime using the Commanded.Event.Upcaster protocol.
By implementing the upcaster protocol you can transform an event before it is used by a consumer. This might be an aggregate, an event handler, or a process manager. Because the upcaster changes the event at runtime, handlers only need to support the latest version. You can also use upcasting to change the type of event.

  
    
  
  Examples


Change the shape of an event by renaming a field:
defimpl Commanded.Event.Upcaster, for: AnEvent do
  def upcast(%AnEvent{} = event, _metadata) do
    %AnEvent{name: name} = event

    %AnEvent{event | first_name: name}
  end
end
Change the type of event by replacing a historical event with a new event:
defimpl Commanded.Event.Upcaster, for: HistoricalEvent do
  def upcast(%HistoricalEvent{} = event, _metadata) do
    %HistoricalEvent{id: id, name: name} = event

    %NewEvent{id: id, name: name}
  end
end

  
    
  
  Reset an EventHandler


An event handler can be reset (using a mix task), it will restart the event store subscription from the configured
start_from. This allow an individual handler to be restart while the app is still running.
You can implement the before_reset/0 callback that will be called before resetting the event handler.
defmodule ExampleHandler do
  use Commanded.Event.Handler,
    application: ExampleApp,
    name: __MODULE__

  require Logger

  alias Commanded.Event.FailureContext

  @impl Commanded.Event.Handler
  def before_reset do
    # Do something
    :ok
  end
end


  

    
Process managers
    

A process manager is responsible for coordinating one or more aggregates. It handles events and dispatches commands in response. You can think of a process manager as the opposite of an aggregate: aggregates handle commands and create events; process managers handle events and create commands. Process managers have state that can be used to track which aggregates are being orchestrated.
Use the Commanded.ProcessManagers.ProcessManager macro in your process manager module and implement the callback functions defined in the behaviour: Commanded.ProcessManagers.ProcessManager.interested?/1, Commanded.ProcessManagers.ProcessManager.handle/2, Commanded.ProcessManagers.ProcessManager.apply/2, and Commanded.ProcessManagers.ProcessManager.error/3.

  
    
  
  interested?/1


The Commanded.ProcessManagers.ProcessManager.interested?/1 function is used to indicate which events the process manager handles. The response is used to route the event to an existing instance or start a new process instance:
	{:start, process_uuid} - create a new instance of the process manager.
	{:start!, process_uuid} - create a new instance of the process manager (strict).
	{:continue, process_uuid} - continue execution of an existing process manager.
	{:continue!, process_uuid} - continue execution of an existing process manager (strict).
	{:stop, process_uuid} - stop an existing process manager, shutdown its
process, and delete its persisted state.
	false - ignore the event.

You can return a list of process identifiers when a single domain event is to be handled by multiple process instances.

  
    
  
  Strict process routing


Using strict routing, with :start! or :continue!, enforces the following validation checks:
	{:start!, process_uuid} - validate process does not already exist.
	{:continue!, process_uuid} - validate process already exists.

If the check fails an error will be passed to the error/3 callback function:
	{:error, {:start!, :process_already_started}}
	{:error, {:continue!, :process_not_started}}

The Commanded.ProcessManagers.ProcessManager.error/3 function can choose to :stop the process or :skip the problematic event.

  
    
  
  handle/2


A handle/2 function can be defined for each :start and :continue tagged event previously specified. It receives the process manager's state and the event to be handled. It must return the commands to be dispatched. This may be none, a single command, or many commands.
The handle/2 function can be omitted if you do not need to dispatch a command and are only mutating the process manager's state.

  
    
  
  apply/2


The Commanded.ProcessManagers.ProcessManager.apply/2 function is used to mutate the process manager's state. It receives the current state and the domain event, and must return the modified state.
This callback function is optional, the default behaviour is to retain the process manager's current state.

  
    
  
  error/3


You can define an Commanded.ProcessManagers.ProcessManager.error/3 callback function to handle any errors or exceptions during event handling or returned by commands dispatched from your process manager. The function is passed the error (e.g. {:error, :failure}), the failed event or command, and a failure context. See Commanded.ProcessManagers.FailureContext for details.
Use pattern matching on the error and/or failed event/command to explicitly handle certain errors, events, or commands. You can choose to retry, skip, ignore, or stop the process manager after a command dispatch error.
The default behaviour, if you don't provide an Commanded.ProcessManagers.ProcessManager.error/3 callback, is to stop the process manager using the exact error reason returned from the event handler function or command dispatch.
The Commanded.ProcessManagers.ProcessManager.error/3 callback function must return one of the following responses depending upon the severity of error and how you choose to handle it:
	{:retry, context} - retry the failed command, provide a context map containing any state passed to subsequent failures. This could be used to count the number of retries, failing after too many attempts.

	{:retry, delay, context} - retry the failed command, after sleeping for the requested delay (in milliseconds). Context is a map as described in {:retry, context} above.

	{:stop, reason} - stop the process manager with the given reason.


For event handling failures, when failure source is an event, you can also return:
	:skip - to skip the problematic event. No commands will be dispatched.

For command dispatch failures, when failure source is a command, you can also return:
	{:continue, commands, context} - continue dispatching the given commands. This allows you to retry the failed command, modify it and retry, drop it, or drop all pending commands by passing an empty list [].

	{:skip, :discard_pending} - discard the failed command and any pending commands.

	{:skip, :continue_pending} - skip the failed command, but continue dispatching any pending commands.



  
    
  
  Supervision


Supervise your process managers to ensure they are restarted on error.
defmodule Bank.Payments.Supervisor do
  use Supervisor

  def start_link(_arg) do
    Supervisor.start_link(__MODULE__, [], name: __MODULE__)
  end

  def init(_arg) do
    Supervisor.init(
      [
        Bank.Payments.TransferMoneyProcessManager
      ],
      strategy: :one_for_one
    )
  end
end

  
    
  
  Supervision caveats


The default error handling strategy is to stop the process manager. When supervised, the process will be restarted and will attempt to handle the same event again, which will likely result in the same error. This could lead to too many restarts of the supervisor, which may eventually cause the application to stop, depending upon your supervision tree and its strategy.
To prevent this you can choose to define the default error handling strategy as :skip to skip over any problematic events.
defmodule ExampleProcessManager do
  use Commanded.ProcessManagers.ProcessManager,
    application: ExampleApp,
    name: __MODULE__

  require Logger

  # By default skip any problematic events
  def error(error, _command_or_event, _failure_context) do
    Logger.error(fn -> "#{__MODULE__} encountered an error: " <> inspect(error) end)

    :skip
  end
end
Alternatively you can define the restart strategy of your process manager as :temporary to prevent it from being restarted on termination. This approach will require manual intervention to fix the stopped process manager, but ensures that it won't miss any events nor crash the application.

  
    
  
  Error handling example


Define an error/3 callback function to determine how to handle errors during event handling and command dispatch.
defmodule ExampleProcessManager do
  use Commanded.ProcessManagers.ProcessManager,
    application: ExampleApp,
    name: "ExampleProcessManager"

  # Stop process manager after three failures
  def error({:error, _failure}, _failed_message, %{context: %{failures: failures}})
    when failures >= 2
  do
    {:stop, :too_many_failures}
  end

  # Retry command, record failure count in context map
  def error({:error, _failure}, _failed_message, %{context: context}) do
    context = Map.update(context, :failures, 1, fn failures -> failures + 1 end)
    {:retry, context}
  end
end
The default behaviour if you don't provide an error/3 callback is to stop the process manager using the same error reason returned from the failed command dispatch.

  
    
  
  Example process manager


defmodule TransferMoneyProcessManager do
  use Commanded.ProcessManagers.ProcessManager,
    application: ExampleApp,
    name: "TransferMoneyProcessManager"

  @derive Jason.Encoder
  defstruct [
    :transfer_uuid,
    :debit_account,
    :credit_account,
    :amount,
    :status
  ]

  # Process routing

  def interested?(%MoneyTransferRequested{transfer_uuid: transfer_uuid}), do: {:start, transfer_uuid}
  def interested?(%MoneyWithdrawn{transfer_uuid: transfer_uuid}), do: {:continue, transfer_uuid}
  def interested?(%MoneyDeposited{transfer_uuid: transfer_uuid}), do: {:stop, transfer_uuid}
  def interested?(_event), do: false

  # Command dispatch

  def handle(%TransferMoneyProcessManager{}, %MoneyTransferRequested{} = event) do
    %MoneyTransferRequested{transfer_uuid: transfer_uuid, debit_account: debit_account, amount: amount} = event

    %WithdrawMoney{account_number: debit_account, transfer_uuid: transfer_uuid, amount: amount}
  end

  def handle(%TransferMoneyProcessManager{} = pm, %MoneyWithdrawn{}) do
    %TransferMoneyProcessManager{transfer_uuid: transfer_uuid, credit_account: credit_account, amount: amount} = pm

    %DepositMoney{account_number: credit_account, transfer_uuid: transfer_uuid, amount: amount}
  end

  # State mutators

  def apply(%TransferMoneyProcessManager{} = transfer, %MoneyTransferRequested{} = event) do
    %MoneyTransferRequested{transfer_uuid: transfer_uuid, debit_account: debit_account, credit_account: credit_account, amount: amount} = event

    %TransferMoneyProcessManager{transfer |
      transfer_uuid: transfer_uuid,
      debit_account: debit_account,
      credit_account: credit_account,
      amount: amount,
      status: :withdraw_money_from_debit_account
    }
  end

  def apply(%TransferMoneyProcessManager{} = transfer, %MoneyWithdrawn{}) do
    %TransferMoneyProcessManager{transfer |
      status: :deposit_money_in_credit_account
    }
  end
end
The name given to the process manager must be unique. This is used when subscribing to events from the event store to track the last seen event and ensure they are only received once.
{:ok, _pid} = TransferMoneyProcessManager.start_link(start_from: :current)
You can choose to start the process router's event store subscription from the :origin, :current position or an exact event number using the start_from option. The default is to use :origin so it will receive all events. You typically use :current when adding a new process manager to an already deployed system containing historical events.
Process manager instance state is persisted to storage after each handled event. This allows the process manager to resume should the host process terminate.

  
    
  
  Configuration options


	consistency - defined as one of either :strong or :eventual (default) for event handling.
	event_timeout - a timeout for event handling to ensure that events are processed in a timely manner without getting stuck.
	idle_timeout - to reduce memory usage you can configure an idle timeout, in milliseconds, after which an inactive process instance will be shutdown.

Refer to the Commanded.ProcessManagers.ProcessManager module docs for more details.


  

    
Supervision
    

Use an OTP supervisor to host your Commanded application, process managers, event handlers, and read model projectors.
defmodule Bank.Supervisor do
  use Supervisor

  def start_link(arg) do
    Supervisor.start_link(__MODULE__, arg, name: __MODULE__)
  end

  @impl true
  def init(_arg) do
    children = [
      # Application
      BankApp,

      # Event handler
      AccountBalanceHandler,

      # Process manager
      TransferMoneyProcessManager,

      # Read model projector
      AccountsProjector,

      # Optionally, provide runtime configuration
      {WelcomeEmailHandler, start_from: :current},
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end
end
Your application should start the supervisor:
defmodule Bank do
  use Application

  def start(_type, _args) do
    Bank.Supervisor.start_link()
  end
end


  

    
Serialization
    


  
    
  
  Default JSON serializer


JSON serialization can be used for event data & metadata, and aggregate and process manager snapshots.
To enable JSON serialization with the included Commanded.Serialization.JsonSerializer module add the jason library to your deps:
def deps do
  [{:jason, "~> 1.2"}]
end
Jason, a pure Elixir JSON library, is used for the actual serialization.
You must derive the Jason.Encoder protocol for all structs you plan on encoding.
defmodule ExampleEvent do
  @derive Jason.Encoder
  defstruct [:name, :datetime]
end
Jason provides an extension point if you need to manually encode your event by using the Jason.Encoder protocol:
defimpl Jason.Encoder, for: Person do
  def encode(%{name: name, age: age}, opts) do
    Jason.Encode.string("#{name} (#{age})", options)
  end
end

  
    
  
  Configuring JSON serialization


JSON serialization configuration depends upon which event store you are using with Commanded. Configure the serializer in config/config.exs, or per environment, as follows.
	Postgres EventStore:
config :my_app, MyApp.EventStore,
  serializer: Commanded.Serialization.JsonSerializer
You can also use the EventStore.JsonSerializer (bytea column type) and EventStore.JsonbSerializer (jsonb column type) serializers which are included in the EventStore library.
config :my_app, MyApp.EventStore,
  serializer: EventStore.JsonSerializer
config :my_app, MyApp.EventStore,
  serializer: EventStore.JsonbSerializer
Note that the two EventStore serializers do not implement the Commanded.Serialization.JsonSerializer decoding protocol.

	EventStoreDB:
config :commanded_extreme_adapter,
  serializer: Commanded.Serialization.JsonSerializer

	In-memory event store:
config :commanded, Commanded.EventStore.Adapters.InMemory,
  serializer: Commanded.Serialization.JsonSerializer



  
    
  
  Decoding event structs


The Commanded.Serialization.JsonSerializer module provides an extension point to allow additional decoding of the deserialized data. This can be used for parsing data into valid types, such as datetime parsing from a string.
The example event below has an implementation of the Commanded.Serialization.JsonDecoder protocol to parse the datetime into a DateTime struct.
defmodule ExampleEvent do
  @derive Jason.Encoder
  defstruct [:name, :datetime]
end

defimpl Commanded.Serialization.JsonDecoder, for: ExampleEvent do
  @doc """
  Parse the datetime included in the event.
  """
  def decode(%ExampleEvent{datetime: datetime} = event) do
    {:ok, dt, _} = DateTime.from_iso8601(datetime)

    %ExampleEvent{event | datetime: dt}
  end
end
The above protocol is only supported by the Commanded.Serialization.JsonSerializer serializer.

  
    
  
  Using an alternative serialization format


You can implement your own serializer module to use an alternative serialization format if preferred.
Configure your own serializer in config/config.exs for the event store you are using.
	PostgreSQL EventStore:
config :my_app, MyApp.EventStore, serializer: MyApp.MessagePackSerializer

	EventStoreDB:
config :my_app, MyApp, serializer: MyApp.MessagePackSerializer

	In-memory event store:
config :my_app, MyApp,
  event_store: [
    adapter: Commanded.EventStore.Adapters.InMemory,
    serializer: MyApp.MessagePackSerializer
  ]


You should not change serialization format once your app has been deployed to production since Commanded will not be able to deserialize any existing events or snapshot data. In this scenario, to change serialization format you would need to also migrate your event store to the new format.

  
    
  
  Customising serialization


To implement custom serialization you could implement your own serializer which extends the behaviour of the existing Commanded.Serialization.JsonSerializer module.
The example below shows how you might compress the serialized data before storage and decompress during deserialization.
defmodule CompressedJsonSerializer do
  def serialize(term) do
    term
    |> Commanded.Serialization.JsonSerializer.serialize()
    |> compress()
  end

  def deserialize(binary, config \\ []) do
    binary
    |> decompress()
    |> Commanded.Serialization.JsonSerializer.deserialize(config)    
  end

  defp compress(term), do: ...
  defp decompress(binary), do: ...
end


  

    
Read model projections
    

Your read model can be built using a Commanded event handler and whichever storage provider you prefer. You can choose to use a SQL or NoSQL database, document store, the filesystem, a full text search index, or any other storage mechanism. You may even use multiple storage providers, optimised for the querying they must support.

  
    
  
  Ecto projections


You can use the Commanded Ecto projections library to build a read model using one of the databases supported by Ecto (PostgreSQL, MySQL, et al).

  
    
  
  Example


defmodule MyApp.ExampleProjector do
  use Commanded.Projections.Ecto,
    application: MyApp.ExampleApp,
    name: "ExampleProjector"

  project %AnEvent{name: name}, _metadata do
    Ecto.Multi.insert(multi, :example_projection, %ExampleProjection{name: name})
  end

  project %AnotherEvent{name: name} do
    Ecto.Multi.insert(multi, :example_projection, %ExampleProjection{name: name})
  end
end

  
    
  
  Consistency guarantee


You will often choose to use :strong consistency for read model projections to ensure that you can query data affected by a dispatched command. In a typical web request using the POST/Redirect/GET pattern you want to ensure the read model is up-to-date before redirecting the user to the modified resource.
By opting in to strong consistency you are guaranteed that an :ok reply from command dispatch indicates all strongly consistent read models will have been updated.
Configure the consistency option in your projector:
defmodule MyApp.ExampleProjector do
  use Commanded.Projections.Ecto,
    application: MyApp.ExampleApp,
    name: "ExampleProjector",
    consistency: :strong
end


  

    
Testing
    

When using ES/CQRS, events are first-class citizens. It's critical to be able to assert that specific events are being emitted. Commanded provides test helpers to simplify your life.

  
    
  
  Setting up your test environment / your tests


Please refer to the Testing your application page on the Wiki for help with configuring your test environment.

  
    
  
  Remember projection_versions when truncating tables


If you rely on your read projections in your tests, remember to truncate the projection_versions table in your truncate_readstore_tables/0 function. Otherwise, your projector will ignore everything but the first projection.
defp truncate_readstore_tables do
    """
    TRUNCATE TABLE
      table1,
      table2,
      table3,
      projection_versions
    RESTART IDENTITY
    CASCADE;
    """
  end

  
    
  
  Asserting that an event is published


Often you'll want to make sure a given event is published. Commanded provides assert_receive_event/3 and assert_receive_event/4 functions in the Commanded.Assertions.EventAssertions module to help with this.
In the first case, we check that any event is received and use the argument as an assertion.
In the second case, we also provide a predicate function that we can use to narrow our search down to a specific event.
import Commanded.Assertions.EventAssertions

test "ensure any event of this type is published" do
  :ok = MyApp.dispatch(%Command{id: 4, date: Date.today})

  assert_receive_event(MyApp, Event, fn event ->
    assert event.id == 4
  end)
end

test "ensure an event is published matching the given predicate" do
  :ok = MyApp.dispatch(%Command{id: 4, date: Date.today})

  assert_receive_event(
    MyApp,
    Event,
    fn event -> event.id == 4 end,
    fn event ->
      assert event.date == Date.today
    end
  )
end

  
    
  
  Waiting for an event


Use the wait_for_event/2 and wait_for_event/3 functions to pause until a specific type of event, or event type matching a given predicate, is received. This can help you deal with eventual consistency in your tests.
import Commanded.Assertions.EventAssertions

test "pause until specific event is published" do
  :ok = BankApp.dispatch(%OpenBankAccount{account_number: "ACC123", initial_balance: 1_000})

  wait_for_event(BankApp, BankAccountOpened, fn opened -> opened.account_number == "ACC123" end)
end

  
    
  
  Correlation


It's a given that when going through CQRS, sometimes many events are part of the same action, either because they are returned from the aggregate together, or because event handlers trigger new commands which generate new events, etc. We will usually want to know, for audit trail purposes, that these events belong together.
For this purpose, Commanded provides assert_correlated/4 which can be used to ensure that specific events have the same correlation_id:
import Commanded.Assertions.EventAssertions

test "make sure two events are correlated" do
  :ok = BankApp.dispatch(%OpenBankAccount{account_number: "ACC123", initial_balance: 1_000})

  assert_correlated(
    BankApp,
    BankAccountOpened, fn opened -> opened.account_number == "ACC123" end,
    InitialAmountDeposited, fn deposited -> deposited.account_number == "ACC123" end
  )
end

  
    
  
  Aggregate state testing


Sometimes it's useful to compare an expected aggregate's state with the previous one. This kind of function should be used
only for testing.
For this purpose, Commanded provides an aggregate_state function which returns the current aggregate state.
import Commanded.Assertions.EventAssertions

alias Commanded.Aggregates.Aggregate

test "make sure aggregate state are what we wanted" do
  account_number = "ACC123"

  :ok = BankApp.dispatch(%OpenBankAccount{account_number: account_number, initial_balance: 1_000})
  :ok = BankApp.dispatch(%WithdrawnMoney{account_number: account_number, amount: 200})

  wait_for_event(BankApp, BankAccountOpened, fn opened -> opened.account_number == "ACC123" end)
  wait_for_event(BankApp, MoneyWithdrawn, fn withdrawn -> withdrawn.balance == 800 end)

  assert Aggregate.aggregate_state(BankApp, BankAccount, account_number) == %BankAccount{
    account_number: account_number,
    balance: 800,
    state: :active
  }
end

  
    
  
  Tests using the event store and read store


To test your application using an event store and read model projection you can take advantage of ExUnit's case template feature to have the databases reset between each test execution. This guarantees that each test starts from a known good state and one test won't affect any other.
First, define a DataCase module which is used to reset the event store and read store databases after each test run using the on_exit/0 callback:
# test/support/data_case.ex
defmodule MyApp.DataCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      import Ecto
      import Ecto.Changeset
      import Ecto.Query
      import Commanded.Assertions.EventAssertions
    end
  end

  setup do
    {:ok, _} = Application.ensure_all_started(:my_app)

    on_exit(fn ->
      :ok = Application.stop(:my_app)

      MyApp.Storage.reset!()
    end)

    :ok
  end
end
The DataCase module uses the following Storage.reset!/0 function to:
	Reset the Postgres EventStore database.
	Truncate the listed tables in the read store database.

Rename table1, table2, and table3 to your own table names and remember to include any new tables when added to your app.
# test/support/storage.ex
defmodule MyApp.Storage do
  @doc """
  Clear the event store and read store databases
  """
  def reset! do
    reset_eventstore()
    reset_readstore()
  end

  defp reset_eventstore do
    config = MyEventStore.config()

    {:ok, conn} = Postgrex.start_link(config)

    EventStore.Storage.Initializer.reset!(conn, config)
  end

  defp reset_readstore do
    config = Application.get_env(:my_app, MyApp.Repo)

    {:ok, conn} = Postgrex.start_link(config)

    Postgrex.query!(conn, truncate_readstore_tables(), [])
  end

  defp truncate_readstore_tables do
    """
    TRUNCATE TABLE
      table1,
      table2,
      table3
    RESTART IDENTITY
    CASCADE;
    """
  end
end
You need to include the test/support files in the test environment Elixir paths by adding the following elixirc_paths/1 function to your app's mix.exs file:
# mix.exs
defmodule MyApp.Mixfile do
  use Mix.Project

  # Include `test/support` files in test environment
  defp elixirc_paths(:test), do: ["lib", "test/support"]
  defp elixirc_paths(_), do: ["lib"]
end
Finally, you can use the MyApp.DataCase case template within any test modules that require access to the databases:
# test/example_test.exs
defmodule MyApp.ExampleTest do
  use MyApp.DataCase

  # Each test will be run against clean read and write databases.
  # After test execution (regardless of success or failure) the databases will be reset.
end
Run your tests using: mix test
When these tests run they will execute against empty event store and read store databases. The caveat with the approach is that the databases will be reset after your tests run; it won't be possible to look at data contained within database after a test fails to debug the failure. The workaround is to temporarily disable the reset operation, run a single failing test, and then you will be able to look at the data. Note that the next time you run any test it will fail because the databases won't have been reset. Subsequent tests will behave normally, assuming you reinstate the reset behaviour.

Using the in-memory event store for testing
You can use the [[InMemoryEventStore]] to test your application. You can also use an ExUnit CaseTemplate to have the in-memory event store restarted between each test run.

Using strong consistency for tests only
You can configure environment specific consistency setting for Commanded event handlers:
# config/config.exs
use Mix.Config

config :my_app, consistency: :eventual
# config/test.exs
use Mix.Config

config :my_app, consistency: :strong
Then read the setting when defining your event handlers and process managers:
defmodule ExampleEventHandler do
  use Commanded.Event.Handler,
    name: __MODULE__,
    consistency: Application.get_env(:my_app, :consistency, :eventual)
end

Given events / When command / Then assert tests
In your test you can append events to the aggregate's event stream to setup its given state. Use Commanded.EventStore.append_to_stream/4 to append events directly to the event store you've configured to use with Commanded. This allows you to configure a different event store for each environment (e.g. in-memory event store for test env).
You need to map your app's domain events to Commanded.EventStore.EventData structs as follows:
causation_id = UUID.uuid4()
correlation_id = UUID.uuid4()

event_data =
  Enum.map(events, fn -> event
    %Commanded.EventStore.EventData{
      causation_id: causation_id,
      correlation_id: correlation_id,
      event_type: Commanded.EventStore.TypeProvider.to_string(event),
      data: event,
      metadata: %{},
    }
  )

{:ok, _} = Commanded.EventStore.append_to_stream(application, stream_uuid, expected_version, event_data)
The stream_uuid will be your aggregate's identity and expected_version is the aggregate version (count of events already appended to its stream, use 0 when creating a new aggregate).
Once you've appended the events, you can dispatch the command via your router. The aggregate process will be started, it'll fetch its events, including those you just appended, and then handle the command.


  

    
Deployment
    

Commanded supports running on a single node, or multiple nodes run as either a distributed Erlang cluster or as multiple single instance nodes.

  
    
  
  Single node deployment


Running your app using Commanded on a single node requires no configuration as local is the default setting.

  
    
  
  Multi node distributed Erlang deployment


To support deployment to a cluster of nodes and distributed Erlang you must configure:
	A registry which supports distributed Erlang. The :global registry provided by Commanded or the Commanded Swarm registry library.
	Phoenix's distributed pub/sub and presence platform to allow process distribution and communication amongst all nodes in the cluster.


  
    
  
  :global registry


Use Erlang's :global name registration facility with distributed Erlang. The global name server starts automatically when a node is started. The registered names are stored in replicated global name tables on every node. Thus, the translation of a name to a pid is fast, as it is always done locally.
Define the :global registry for your application:
defmodule MyApp.Application do
  use Commanded.Application,
    otp_app: :my_app,
    event_store: [
      adapter: Commanded.EventStore.Adapters.EventStore,
      event_store: MyApp.EventStore
    ],
    registry: :global
end
Or configure your application to use the :global registry in config:
# config/config.exs
config :my_app, MyApp.Application, registry: :global
Note that when clusters are formed dynamically (e.g. using libcluster]), the typical sequence of events is that first all nodes will start all processes, then the cluster is formed and :global will kill off duplicate names. This is ugly in the logs but expected; it also means that if your supervisor's :max_restarts is too low - lower than the number of event handlers/projectors you start - it will immediately exit and if that was your
application supervisor, your app gets shutdown. The solution is simple: keep :max_restarts above the number of event handlers you start under your supervisor and the transition from no cluster to cluster will be clean. 

  
    
  
  Commanded Swarm registry


Add commanded_swarm_registry to your list of dependencies in mix.exs:
def deps do
  [{:commanded_swarm_registry, "~> 1.0"}]
end
Fetch mix dependencies:
$ mix deps.get

Configure your application to use the Swarm registry:
# config/config.exs
config :my_app, MyApp.Application, registry: Commanded.Registration.SwarmRegistry
This uses the Swarm to distribute processes amongst the available nodes in the cluster.

  
    
  
  Phoenix pub/sub


First, add it as a dependency to your project's mix.exs file:
defp deps do
  [{:phoenix_pubsub, "~> 2.0"}]
end
Fetch mix dependencies and configure the pubsub settings for your application in the environment config file:
# config/config.exs
config :my_app, MyApp.Application,
  pubsub: [
    phoenix_pubsub: [
      adapter: Phoenix.PubSub.PG2,
      pool_size: 1
    ]
  ]
The PG2 adapter is preferable for cluster deployment since it is provided by Erlang and requires no further dependencies.

  
    
  
  EventStore


If using PostgreSQL-based Elixir EventStore please also refer to its documentation about running on a clustering of nodes.

  
    
  
  Multi-node, but not distributed Erlang deployment


Running multiple nodes, but choosing not to connect the nodes together to form a distributed Erlang cluster, requires that you use the local registry and Phoenix's pub/sub library with its Redis adapter.
You must install and use Phoenix's pub/sub library, as described above.
Since the nodes aren't connected, you are required to use the Redis adapter as a way of communicating between the nodes. Therefore you will need to host a Redis instance for use by your app.
# config/config.exs
config :my_app, MyApp.Application,
  registry: :local,
  pubsub: [
    phoenix_pubsub: [
      adapter: Phoenix.PubSub.Redis,
      host: "localhost",
      port: 6379,
      node_name: "localhost"
    ]
  ]


  

    
Upgrade guide v0.19.x to v1.0
    

In v0.19.x and earlier a singleton Commanded application was used to host aggregates and support event handlers, process managers, and other infrastructure processes. For v1.0, support for multiple Commanded apps was added allowing you to define and use more than one Commanded app and to control its lifecycle. Follow the following upgrade advice to migrate your application to Commanded v1.0.

  
    
  
  Commanded application


First you must define a Commanded application:
defmodule MyApp.Application do
  use Commanded.Application, otp_app: :my_app

  router(MyApp.Router)
end
The application needs to be configured. You can do so in application config:
# config/config.exs
config :my_app, MyApp.Application,
  event_store: [
    adapter: Commanded.EventStore.Adapters.EventStore,
    event_store: MyApp.EventStore
  ],
  pubsub: :local,
  registry: :local
Alternatively, you can include the event store, pubsub, and registry config when defining the application:
defmodule MyApp.Application do
  use Commanded.Application,
    otp_app: :my_app,
    event_store: [
      adapter: Commanded.EventStore.Adapters.EventStore,
      event_store: MyApp.EventStore
    ],
    pubsub: :local,
    registry: :local

  router(MyApp.Router)
end

  
    
  
  Event store configuration


To use Commanded v1.0 you must also upgrade the adapter for the event store you are using.
Postgres EventStore
Use the commanded_eventstore_adapter and follow the Getting started guide.
In brief, the upgrade requires you to:
	Define an event store module for your Commanded application:
 defmodule MyApp.EventStore do
   use EventStore, otp_app: :my_app
 end

	Add a config entry containing the PostgreSQL database connection details for your event store module to each environment's mix config file (e.g. config/dev.exs):
 config :my_app, MyApp.EventStore,
   serializer: Commanded.Serialization.JsonSerializer,
   username: "postgres",
   password: "postgres",
   database: "eventstore",
   hostname: "localhost"

	Configure the Commanded application to use the event store:
 defmodule MyApp.Application do
   use Commanded.Application,
     otp_app: :my_app,
     event_store: [
       adapter: Commanded.EventStore.Adapters.EventStore,
       event_store: MyApp.EventStore
     ]
 end


Event Store
Use the commanded_extreme_adapter and follow the Getting started guide.

  
    
  
  Supervision


The Commanded application must be included and started somewhere in your app's supervision tree. You can include it within the top level application supervisor:
defmodule MyApp do
  use Application

  def start(_type, _args) do
    children = [
      MyApp.Application
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end  
end

  
    
  
  Command dispatch


Previously commands were always dispatched via a router module. Now, you dispatch commands using your Commanded application:
:ok = MyApp.Application.dispatch(%DoSomething{..})
The above should be a straightforward find within your source code to replace the router module (e.g. MyApp.Router) with the new application module (e.g. MyApp.Application).
Optionally, you can dispatch a command using an existing router by configuring the router with the Commanded application:
defmodule MyApp.Router do
  use Commanded.Commands.Router, application: MyApp.Application
end

:ok = MyApp.Router.dispatch(%DoSomething{..})
It is also possible to specify the application during command dispatch:
:ok = MyApp.Router.dispatch(%DoSomething{..}, application: MyApp.Application)

  
    
  
  Event handlers and process managers


Each handler, read model projections, and process manager needs to be provided with the additional application option:
defmodule MyApp.ExampleHandler do
  use Commanded.Event.Handler,
    application: MyApp.Application,
    name: "MyApp.ExampleHandler"
end

  
    
  
  Upgrade Commanded supporting libraries


You will need to upgrade to the v1.0 versions of any Commanded related library used by your application, such as the event store adapter and Ecto read model projections.


  

    
Commanded 
    



      
Use Commanded to build your own Elixir applications following the CQRS/ES pattern.
Provides support for:
	Defining applications.
	Command registration and dispatch.
	Hosting and delegation to aggregates.
	Event handling.
	Long running process managers.

Use Commanded with one of the following event stores for persistence:
	Elixir EventStore using Postgres for persistence
	EventStoreDB

Please check the Getting Started and Usage guides to learn more.

    Commanded.Application - Commanded v1.4.8
    
    

    


  
  

    
Commanded.Application behaviour
    



      
Defines a Commanded application.
The application expects at least an :otp_app option to be specified. It
should point to an OTP application that has the application configuration.
For example, the application:
defmodule MyApp.Application do
  use Commanded.Application, otp_app: :my_app

  router(MyApp.Router)
end
Could be configured with:
# config/config.exs
config :my_app, MyApp.Application
  event_store: [
    adapter: Commanded.EventStore.Adapters.EventStore,
    event_store: MyApp.EventStore
  ],
  pubsub: :local,
  registry: :local
Alternatively, you can include the configuration when defining the
application:
defmodule MyApp.Application do
  use Commanded.Application,
    otp_app: :my_app,
    event_store: [
      adapter: Commanded.EventStore.Adapters.EventStore,
      event_store: MyApp.EventStore
    ],
    pubsub: :local,
    registry: :local

  router(MyApp.Router)
end
A Commanded application must be started before it can be used:
{:ok, _pid} = MyApp.Application.start_link()
Instead of starting the application manually, you should use a
Supervisor.

  
    
  
  Supervision


Use a supervisor to start your Commanded application:
Supervisor.start_link([
  MyApp.Application
], strategy: :one_for_one)

  
    
  
  Command routing


Commanded applications are also composite routers allowing you to include
one or more routers within an application.

  
    
  
  Example


defmodule MyApp.Application do
  use Commanded.Application, otp_app: :my_app

  router(MyApp.Accounts.Router)
  router(MyApp.Billing.Router)
  router(MyApp.Notifications.Router)
end
See Commanded.Commands.CompositeRouter for details.

  
    
  
  Command dispatch


Once a router has been configured you can dispatch a command via the
application:
:ok = MyApp.dispatch(command, opts)
See dispatch/1 and dispatch/2 for details.

  
    
  
  Dynamic named applications


An application can be provided with a name as an option to start_link/1.
This can be used to start the same application multiple times, each using its
own separately configured and isolated event store. Each application must be
started with a unique name.
Multiple instances of the same event handler or process manager can be
started by referring to a started application by its name. The event store
operations can also be scoped to an application by referring to its name.

  
    
  
  Example


Start an application process for each tenant in a multi-tenanted app,
guaranteeing that the data and processing remains isolated between tenants.
for tenant <- [:tenant1, :tenant2, :tenant3] do
  {:ok, _app} = MyApp.Application.start_link(name: tenant)
end
Typically you would start the applications using a supervisor:
children =
  for tenant <- [:tenant1, :tenant2, :tenant3] do
    {MyApp.Application, name: tenant}
  end

Supervisor.start_link(children, strategy: :one_for_one)
To dispatch a command you must provide the application name:
:ok = MyApp.Application.dispatch(command, application: :tenant1)

  
    
  
  Default dispatch options


An application can be configured with default command dispatch options such as
:consistency, :timeout, and :returning. Any defaults will be used
unless overridden by options provided to the dispatch function.
defmodule MyApp.Application do
  use Commanded.Application,
    otp_app: :my_app,
    default_dispatch_opts: [
      consistency: :eventual,
      returning: :aggregate_version
    ]
end
See the Commanded.Commands.Router module for more details about the
supported options.

  
    
  
  Telemetry


	[:commanded, :application, :dispatch, :start]
	Description: Emitted when an application starts dispatching a command
	Measurements: %{system_time: integer()}
	Metadata: %{application: Commanded.Application.t(), execution_context: Commanded.Aggregates.ExecutionContext.t()}


	[:commanded, :application, :dispatch, :stop]
	Description: Emitted when an application stops dispatching a command
	Measurements: %{duration: non_neg_integer()}
	Metadata: %{application: Commanded.Application.t(), execution_context: Commanded.Aggregates.ExecutionContext.t(), error: nil | any()}

Commanded.UUID 
    



      
Generate and format random, version 4 UUIDs.

Commanded.Aggregate.Multi 
    



      
Use Commanded.Aggregate.Multi to generate multiple events from a single
command.
This can be useful when you want to emit multiple events that depend upon the
aggregate state being updated.

  
    
  
  Example


In the example below, money is withdrawn from the bank account and the
updated balance is used to check whether the account is overdrawn.
defmodule BankAccount do
  alias Commanded.Aggregate.Multi

  defstruct [:account_number, :state, balance: 0]

  def withdraw(
    %BankAccount{state: :active} = account,
    %WithdrawMoney{amount: amount})
    when is_number(amount) and amount > 0
  do
    account
    |> Multi.new()
    |> Multi.execute(&withdraw_money(&1, amount))
    |> Multi.execute(&check_balance/1)
  end

  defp withdraw_money(%BankAccount{account_number: account_number, balance: balance}, amount) do
    %MoneyWithdrawn{
      account_number: account_number,
      amount: amount,
      balance: balance - amount
    }
  end

  defp check_balance(%BankAccount{account_number: account_number, balance: balance})
    when balance < 0
  do
    %AccountOverdrawn{account_number: account_number, balance: balance}
  end
  defp check_balance(%BankAccount{}), do: []
end

Commanded.Aggregates.Aggregate 
    



      
Aggregate is a GenServer process used to provide access to an
instance of an event sourced aggregate.
It allows execution of commands against an aggregate instance, and handles
persistence of created events to the configured event store. Concurrent
commands sent to an aggregate instance are serialized and executed in the
order received.
The Commanded.Commands.Router module will locate, or start, an aggregate
instance when a command is dispatched. By default, an aggregate process will
run indefinitely once started. Its lifespan may be controlled by using the
Commanded.Aggregates.AggregateLifespan behaviour.

  
    
  
  Snapshotting


You can configure state snapshots for an aggregate in config. By default
snapshots are not taken for an aggregate. The following options are
available to enable snapshots:
	snapshot_every - snapshot aggregate state every so many events. Use
nil to disable snapshotting, or exclude the configuration entirely.

	snapshot_version - a non-negative integer indicating the version of
the aggregate state snapshot. Incrementing this version forces any
earlier recorded snapshots to be ignored when rebuilding aggregate
state.



  
    
  
  Example


In config/config.exs enable snapshots for MyApp.ExampleAggregate after
every ten events:
config :my_app, MyApp.Application,
  snapshotting: %{
    MyApp.ExampleAggregate => [
      snapshot_every: 10,
      snapshot_version: 1
    ]
  }

  
    
  
  Telemetry


	[:commanded, :aggregate, :execute, :start]
	Description: Emitted when an aggregate starts executing a command
	Measurements: %{system_time: integer()}
	Metadata: %{application: Commanded.Application.t(), aggregate_uuid: String.t(), aggregate_state: struct(), aggregate_version: non_neg_integer(), caller: pid(), execution_context: Commanded.Aggregates.ExecutionContext.t()}


	[:commanded, :aggregate, :execute, :stop]
	Description: Emitted when an aggregate stops executing a command
	Measurements: %{duration: non_neg_integer()}
	Metadata: %{application: Commanded.Application.t(), aggregate_uuid: String.t(), aggregate_state: struct(), aggregate_version: non_neg_integer(), caller: pid(), execution_context: Commanded.Aggregates.ExecutionContext.t(), events: [map()], error: nil | any()}


	[:commanded, :aggregate, :execute, :exception]
	Description: Emitted when an aggregate raises an exception
	Measurements: %{duration: non_neg_integer()}
	Metadata: %{application: Commanded.Application.t(), aggregate_uuid: String.t(), aggregate_state: struct(), aggregate_version: non_neg_integer(), caller: pid(), execution_context: Commanded.Aggregates.ExecutionContext.t(), kind: :throw | :error | :exit, reason: any(), stacktrace: list()}

Commanded.Aggregates.AggregateLifespan behaviour
    



      
The Commanded.Aggregates.AggregateLifespan behaviour is used to control the
aggregate GenServer process lifespan.
By default an aggregate instance process will run indefinitely once started.
You can change this default by implementing the
Commanded.Aggregates.AggregateLifespan behaviour in a module and configuring
it in your router.
After a command successfully executes, and creates at least one domain event,
the after_event/1 function is called passing the last created event.
When a command is successfully handled but results in no domain events (by
returning nil or an empty list []), the command struct is passed to the
after_command/1 function.
Finally, if there is an error executing the command, the error reason is
passed to the after_error/1 function.
For all the above, the returned inactivity timeout value is used to shutdown
the aggregate process if no other messages are received.

  
    
  
  Supported return values


	Non-negative integer - specify an inactivity timeout, in millisconds.
	:infinity - prevent the aggregate instance from shutting down.
	:hibernate - send the process into hibernation.
	:stop - immediately shutdown the aggregate process with a :normal exit
reason.
	{:stop, reason} - immediately shutdown the aggregate process with the
given reason.


  
    
  
  Hibernation


A hibernated process will continue its loop once a message is in its message
queue. Hibernating an aggregate causes garbage collection and minimises the
memory used by the process. Hibernating should not be used aggressively as too
much time could be spent garbage collecting.

  
    
  
  Example


Define a module that implements the Commanded.Aggregates.AggregateLifespan
behaviour:
defmodule BankAccountLifespan do
  @behaviour Commanded.Aggregates.AggregateLifespan

  def after_event(%MoneyDeposited{}), do: :timer.hours(1)
  def after_event(%BankAccountClosed{}), do: :stop
  def after_event(_event), do: :infinity

  def after_command(%CloseAccount{}), do: :stop
  def after_command(_command), do: :infinity

  def after_error(:invalid_initial_balance), do: :timer.minutes(5)
  def after_error(_error), do: :stop
end
Then specify the module as the lifespan option when registering the
applicable commands in your router:
defmodule BankRouter do
  use Commanded.Commands.Router

  dispatch [OpenAccount, CloseAccount],
    to: BankAccount,
    lifespan: BankAccountLifespan,
    identity: :account_number
end

Commanded.Aggregates.DefaultLifespan 
    



      
The default implementation of the Commanded.Aggregates.AggregateLifespan
behaviour.
It will ensure that an aggregate instance process runs indefinitely once
started, unless an exception is encountered.

Defines the arguments used to execute a command for an aggregate.
The available options are:
	command - the command to execute, typically a struct
(e.g. %OpenBankAccount{...}).

	retry_attempts - the number of retries permitted if an
{:error, :wrong_expected_version} is encountered when appending events.

	causation_id - the UUID assigned to the dispatched command.

	correlation_id - a UUID used to correlate related commands/events.

	metadata - a map of key/value pairs containing the metadata to be
associated with all events created by the command.

	handler - the module that handles the command. It may be either the
aggregate module itself or a separate command handler module.

	function - the name of the function, as an atom, that handles the command.
The default value is :execute, used to support command dispatch directly
to the aggregate module. For command handlers the :handle function is
used.

	before_execute - the name of the function, as an atom, that prepares the
command before execution, called just before function. The default value
is nil, disabling it. It should return :ok on success or {:error, any()}
to cancel the dispatch.

	lifespan - a module implementing the Commanded.Aggregates.AggregateLifespan
behaviour to control the aggregate instance process lifespan. The default
value, Commanded.Aggregates.DefaultLifespan, keeps the process running
indefinitely.



      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Commanded.Aggregates.Supervisor - Commanded v1.4.8
    
    

    


  
  

    
Commanded.Aggregates.Supervisor 
    



      
Supervises Commanded.Aggregates.Aggregate instance processes.

Commanded.Commands.CompositeRouter 
    



      
Composite router allows you to combine multiple router modules into a single
router able to dispatch any registered command from an included child router.
One example usage is to define a router per context and then combine each
context's router into a single top-level composite app router used for all
command dispatching.

  
    
  
  Example


Define a composite router module which imports the commands from each included
router:
defmodule Bank.AppRouter do
  use Commanded.Commands.CompositeRouter

  router(Bank.Accounts.Router)
  router(Bank.MoneyTransfer.Router)
end
  One or more routers or composite routers can be included in a
  Commanded.Application  since it is also a composite router:
defmodule BankApp do
  use Commanded.Application

  router(Bank.AppRouter)
end
You can dispatch a command via the application which will then be routed to
the associated child router:
command = %OpenAccount{account_number: "ACC123", initial_balance: 1_000}

:ok = BankApp.dispatch(command)
Or via the composite router itself, specifying the application:
:ok = Bank.AppRouter.dispatch(command, application: BankApp)
A composite router can include composite routers.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Commanded.Commands.ExecutionResult - Commanded v1.4.8
    
    

    


  
  

    
Commanded.Commands.ExecutionResult 
    



      
Contains the aggregate, events, and metadata created by a successfully
executed command.
The available fields are:
	aggregate_uuid - identity of the aggregate instance.

	aggregate_state - resultant state of the aggregate after executing
the command.

	aggregate_version - resultant version of the aggregate after executing
the command.

	events - a list of the created events, it may be an empty list.

	metadata - an map containing the metadata associated with the command
dispatch.

Defines the behaviour a command handler module must implement to support command dispatch.

  
    
  
  Example


An open account handler that delegates to a bank account aggregate:
defmodule OpenAccountHandler do
  @behaviour Commanded.Commands.Handler

  def handle(%BankAccount{} = aggregate, %OpenAccount{} = command) do
    %OpenAccount{account_number: account_number, initial_balance: initial_balance} = command

    BankAccount.open_account(aggregate, account_number, initial_balance)
  end
end

Command routing macro to allow configuration of each command to its command handler.

  
    
  
  Example


Define a router module which uses Commanded.Commands.Router and configures
available commands to dispatch:
defmodule BankRouter do
  use Commanded.Commands.Router

  dispatch OpenAccount,
    to: OpenAccountHandler,
    aggregate: BankAccount,
    identity: :account_number
end
The to option determines which module receives the command being dispatched.
This command handler module must implement a handle/2 function. It receives
the aggregate's state and the command to execute. Usually the command handler
module will forward the command to the aggregate.
Once configured, you can either dispatch a command by using the module and
specifying the application:
command = %OpenAccount{account_number: "ACC123", initial_balance: 1_000}

:ok = BankRouter.dispatch(command, application: BankApp)
Or, more simply, you should include the router module in your application:
defmodule BankApp do
  use Commanded.Application, otp_app: :my_app

  router MyApp.Router
end
Then dispatch commands using the app:
command = %OpenAccount{account_number: "ACC123", initial_balance: 1_000}

:ok = BankApp.dispatch(command)

  
    
  
  Dispatch command directly to an aggregate


You can route a command directly to an aggregate, without requiring an
intermediate command handler.

  
    
  
  Example


defmodule BankRouter do
  use Commanded.Commands.Router

  # Will route to `BankAccount.open_account/2`
  dispatch OpenAccount, to: BankAccount, identity: :account_number
end
By default, you must define an execute/2 function on the aggregate module, which will be
called with the aggregate's state and the command to execute. Using this approach, you must
create an execute/2 clause that pattern-matches on each command that the aggregate should
handle.
Alternatively, you may specify the name of a function (also receiving both the aggregate state
and the command) on your aggregate module to which the command will be dispatched:

  
    
  
  Example


defmodule BankRouter do
  use Commanded.Commands.Router

  # Will route to `BankAccount.open_account/2`
  dispatch OpenAccount, to: BankAccount, function: :open_account, identity: :account_number
end

  
    
  
  Define aggregate identity


You can define the identity field for an aggregate once using the identify macro.
The configured identity will be used for all commands registered to the aggregate,
unless overridden by a command registration.

  
    
  
  Example


defmodule BankRouter do
  use Commanded.Commands.Router

  identify BankAccount,
    by: :account_number,
    prefix: "bank-account-"

  dispatch OpenAccount, to: BankAccount
end
An optional identity prefix can be used to distinguish between different
aggregates that  would otherwise share the same identity. As an example you
might have a User and a UserPreferences aggregate that you wish
to share the same identity. In this scenario you should specify a prefix
for each aggregate (e.g. "user-" and "user-preference-").
The prefix is used as the stream identity when appending and reading the
aggregate's events: "<identity_prefix><aggregate_uuid>". It can be a string or
a zero arity function returning a string.

  
    
  
  Consistency


You can choose the consistency guarantee when dispatching a command. The
available options are:
	:eventual (default) - don't block command dispatch while waiting for
event handlers
:ok = BankApp.dispatch(command)
:ok = BankApp.dispatch(command, consistency: :eventual)

	:strong - block command dispatch until all strongly
consistent event handlers and process managers have successfully processed
all events created by the command.
Use this when you have event handlers that update read models you need to
query immediately after dispatching the command.
:ok = BankApp.dispatch(command, consistency: :strong)

	Provide an explicit list of event handler and process manager modules (or
their configured names), containing only those handlers you'd like to wait
for. No other handlers will be awaited on, regardless of their own
configured consistency setting.
:ok = BankApp.dispatch(command, consistency: [ExampleHandler, AnotherHandler])
:ok = BankApp.dispatch(command, consistency: ["ExampleHandler", "AnotherHandler"])
Note you cannot opt-in to strong consistency for a handler that has been
configured as eventually consistent.



  
    
  
  Dispatch return


By default a successful command dispatch will return :ok. You can change
this behaviour by specifying a returning option.
The supported options are:
	:aggregate_state - to return the update aggregate state.

	:aggregate_version - to return only the aggregate version.

	:events - to return the resultant domain events. An empty list will be
returned if no events were produced.

	:execution_result - to return a Commanded.Commands.ExecutionResult
struct containing the aggregate's identity, state, version, and any events
produced from the command along with their associated metadata.

	false - don't return anything except an :ok.



  
    
  
  Aggregate state


Return the updated aggregate state as part of the dispatch result:
{:ok, %BankAccount{}} = BankApp.dispatch(command, returning: :aggregate_state)
This is useful when you want to immediately return fields from the aggregate's
state without requiring an read model projection and waiting for the event(s)
to be projected. It may also be appropriate to use this feature for unit
tests.
However, be warned that tightly coupling an aggregate's state with read
requests may be harmful. It's why CQRS enforces the separation of reads from
writes by defining two separate and specialised models.

  
    
  
  Aggregate version


You can optionally choose to return the aggregate's version as part of the
dispatch result:
{:ok, aggregate_version} = BankApp.dispatch(command, returning: :aggregate_version)
This is useful when you need to wait for an event handler, such as a read model
projection, to be up-to-date before querying its data.

  
    
  
  Execution results


You can also choose to return the execution result as part of the dispatch
result:
alias Commanded.Commands.ExecutionResult

{:ok, %ExecutionResult{} = result} = BankApp.dispatch(command, returning: :execution_result)
Or by setting the default_dispatch_return in your application config file:
# config/config.exs
config :commanded, default_dispatch_return: :execution_result
Use the execution result struct to get information from the events produced
from the command.

  
    
  
  Metadata


You can associate metadata with all events created by the command.
Supply a map containing key/value pairs comprising the metadata:
:ok = BankApp.dispatch(command, metadata: %{"ip_address" => "127.0.0.1"})

Data related to an event handling failure.
The available fields are:
	:application - the associated Commanded.Application.

	:handler_name - the name of the event handler.

	:handler_state - optional event handler state.

	:context - a map that is passed between each failure. Use it to store
any transient state between